

Саратовский государственный университет имени Н.Г.Чернышевского  
Кафедра полимеров на базе ООО «АКРИПОЛ»

С.Л. Шмаков

Сборник программ МАХІМА  
по статистической физике  
макромолекул

Саратов, 2017

УДК [539.199:531.19]: 519.67(075.8)

Шмаков С.Л. Сборник программ МАХІМА по статистической физике макромолекул: Учебно-методич. пособие. – Саратов: Саратовск. госуниверситет, 2016. – 63 с.

В пособии собраны программы (документы) свободно распространяемого пакета компьютерной математики МАХІМА, оптимизирующие решение задач статистической физики макромолекул. Отдельные темы названной дисциплины излагаются с автоматизированным выводом формул. Программы открыты для доработки.

Пособие предназначено для студентов Института химии СГУ, направление «Химия», профиль «Химия низко- и высокомолекулярных соединений», бакалавриат, дисциплина «Компьютерное моделирование полимерных систем». Может использоваться при чтении лекций с проектором и ноутбуком, а также для самостоятельной работы студентов. Также пособие будет полезно самостоятельно осваивающим статистическую физику макромолекул, магистрантам, аспирантам и молодым специалистам.

*Учебно-методическое пособие рекомендовано учебно-методической комиссией Института химии Саратовского государственного университета для бакалавров, магистрантов и аспирантов, обучающихся по направлению подготовки «Химия»*

## Предисловие

Полноценное овладение статистической физикой макромолекул, а тем более теоретическая работа в данной области, требует свободного владения сложным математическим аппаратом, доведения до автоматизма многих действий, иначе будет ускользать физический смысл. Вместе с тем, наличия таких умений и навыков не предполагается у бакалавров, специализирующихся по кафедре полимеров на базе ООО «АКРИПОЛ», ввиду сокращения часов по математике и других негативных факторов. Выходом из положения может служить передача упомянутого автоматизма настоящему автомату, то есть компьютеру, оснащённому соответствующей программой, выполняющей как числовые расчёты, так и аналитические преобразования выражений.

Из целого ряда универсальных математических программ нами была выбрана Maxima — свободно распространяемая программа-ядро с оболочкой для Windows — приложением wxMaxima. Интерфейс последней, конечно, уступает графическому интерфейсу коммерческих программ, однако это вполне компенсируется доступностью: любой студент может установить Maxima на свой компьютер, и мы рекомендуем ему это сделать.

Настоящий сборник ориентирован на специальный курс «Компьютерное моделирование полимерных систем», который автор проводит со студентами IV курса Института химии СГУ, профиль «химия низко- и высокомолекулярных соединений». Под «программой Maxima» мы понимаем совокупность последовательных ячеек (строк) с операторами, взаимосвязанных по существу и направленных на получение определённого результата, независимо от того, оформлена ли данная совокупность программным блоком формально. Создание (включая отладку) таких программ есть не что иное, как программирование, поэтому термин «документ Maxima» нами здесь не используется, тем более что отлаженную последовательность операторов не составляет труда оформить в виде «официального» программного блока.

Конечно, компьютер не может полностью заменить человека. В частности, Maxima «не понимает», в каком именно виде оператор хочет получить от неё результат. Например, переменные в произведении упорядочиваются по алфавиту, а не по физическому смыслу. Иногда желаемой формы итогового выражения можно добиться лишь методом проб и ошибок, перебрав разные способы упрощения выражения, но чаще всего приходится переписывать его вручную, соблюдая неписанные правила компоновки формул в физике. Этому студентов тоже надо научить.

Надеемся, что настоящий сборник окажется полезным и тем, кто использует в своей практике другие математические пакеты, включая коммерческие; бакалаврам, магистрантам и аспирантам, специализирующимся по физике и химии полимеров. Доработка приведённых в сборнике программ, создание и отладка новых приветствуются.

Автор благодарен проф. Кледину В.И. за обсуждение материалов и ценные советы.

## 1. Средние величины в моделях идеальной полимерной цепи

Модель идеальной макромолекулы представляет собой цепочку из бестелесных звеньев, каждое из которых соединено с ближайшими соседями, но не взаимодействует с другими, кроме этих соседей, звеньями той же или других макромолекул, как и с растворителем. Благодаря такому пренебрежению расчёты в идеальной модели сильно упрощаются и во многих случаях могут быть выполнены аналитически.

Наиболее простой топологией макромолекулы является линейная:  $N$  жёстких сегментов одинаковой длины  $a$  каждый шарнирно соединены в линейную цепочку<sup>1</sup>. Мгновенная конформация такой цепи описывается координатами всех сегментов и в общем характеризуется вектором  $\mathbf{R}$ , соединяющим её концы друг с другом. Простейшей характеристикой среднего размера макромолекулы является средний квадрат расстояния между концами  $\langle \mathbf{R}^2 \rangle$ . Очевидно, любое значение любой нечётной степени  $\mathbf{R}$ , сохраняющей знак, будет встречаться равно часто, как и противоположное ему ( $-\mathbf{R}$ ) в силу симметрии пространства, поэтому среднее этой степени будет нулевым.

Пусть  $i$ -й сегмент представлен вектором  $\mathbf{u}_i$ , тогда

$$\mathbf{R} = \sum_{i=1}^N \mathbf{u}_i.$$

Квадрат  $\mathbf{R}$  — это скалярное произведение на самоё себя:

$$\begin{aligned} \langle \mathbf{R}^2 \rangle &= \left\langle \left( \sum_{i=1}^N \mathbf{u}_i \right)^2 \right\rangle = \sum_{i=1}^N \langle \mathbf{u}_i^2 \rangle + 2 \sum_{1 \leq i < j \leq N} \sum \langle \mathbf{u}_i \mathbf{u}_j \rangle \\ &= Na^2 + 2a^2 \sum_{1 \leq i < j \leq N} \sum \langle \cos \vartheta_{ij} \rangle, \end{aligned}$$

где  $\vartheta_{ij}$  — угол между сегментами  $\mathbf{u}_i$  и  $\mathbf{u}_j$ .

Двойная сумма средних косинусов зависит от корреляции положения векторов  $\mathbf{u}_i$ . Рассмотрим несколько вариантов этой корреляции.

### 1.1. Свободно-сочленённая цепь

Направления даже соседних сегментов никак не скоррелированы (а несоседних — тем более), и угол между любыми  $\mathbf{u}_i$  и  $\mathbf{u}_j$  с равной вероятностью принимает любое значение от 0 до  $2\pi$  (или до  $\pi$ , если брать наименьший), поэтому  $\langle \cos \vartheta_{ij} \rangle = 0$ . Отсюда

<sup>1</sup> Часто длину сегмента обозначают  $l$ , но в листингах MAXIMA её легко спутать с единицей. Обычно в таких случаях заменяют строчную букву на соответствующую прописную, но в данном случае буквой  $L$  обозначается контурная длина цепи, поэтому для длины сегмента здесь выбрана буква  $a$ .

$$\langle R^2 \rangle_{free} = Na^2.$$

Заметим, что контурная длина цепи (т.е. длина полностью растянутой цепи) равна  $L = Na$ .

## 1.2. Средний радиус инерции

Средний квадрат радиуса инерции клубка  $s$  определяется формулой

$$\langle s^2 \rangle = \frac{1}{N^2} \sum_{i=2}^N \sum_{j=1}^{i-1} \langle r_{ij}^2 \rangle$$

где  $\mathbf{r}_{ij}$  — вектор, соединяющий звенья  $i$  и  $j$ . Рассматривая субцепь  $i-j$  как отдельную макромолекулу с  $N = i - j$ , запишем

$$\langle r_{ij}^2 \rangle = (i - j)a^2.$$

Проводим расчёт прямым суммированием:

```
(%i1) (i-j)*a^2$
(%i2) sum(%,j,1,i-1),simpsum;
(%o2) a^2*((i-1)i - (i+(i-1)^2-1)/2)
.
(%i3) sum(%,i,2,N),simpsum;
(%o3) a^2*(2*N^3+3*N^2+N)/12 - (N^2+N)/4
.
(%i4) %/N^2,expand;
(%o4) a^2*N/6 - a^2/(6*N)
.
(%i5) N*limit(%/N,N,inf);
(%o5) a^2*N/6
```

Как видим, в строке %o4 стоят два слагаемых, второе из которых становится несоизмеримо малым по сравнению с первым с ростом  $N$ . Делим всё выражение на  $N$ , чтобы первое слагаемое стало постоянным, применяем предел  $N \rightarrow \infty$  (второе слагаемое исчезает), и снова умножаем на  $N$ . Такой приём применяется, когда обычный предел даёт  $\pm\infty$ .

При  $N \gg 1$  также допустимо заменить суммирование интегрированием:

```

(%i6) integrate(%o1,j,1,i-1);
(%o6) a^2 * ( (i^2-1)/2 - (2*i-1)/2 )
(%i7) integrate(%i,i,2,N);
(%o7) a^2 * ( (N^3-3*N^2)/6 + 2/3 )
(%i8) %/N^2,factor;
(%o8) a^2 * (N-2)^2 * (N+1) / (6 * N^2)
(%i9) N*limit(%/N,N,inf);
(%o9) a^2 * N / 6

```

Результат, как видим, получается один и тот же:

$$\langle s^2 \rangle = \frac{Na^2}{6} = \frac{\langle R^2 \rangle}{6}.$$

### 1.3. Цепь с фиксированным валентным углом

Здесь общий угол  $\vartheta$  между соседними сегментами распадается на валентный угол  $\theta$  (удобнее за собственно валентный принять  $\pi-\theta$ ), который фиксирован, и торсионный угол  $\phi$ , с равной вероятностью принимающий любые значения из диапазона  $[0, 2\pi]$  — свободное вращение вокруг ординарной связи. Углы  $\theta$  и  $\phi$  можно уподобить географическим широте и долготе. Тогда

$$\langle R^2 \rangle_{\theta} = Na^2 + 2a^2 \sum_{i=1}^N \sum_{j=1}^{N-i} \langle \cos \theta_{i,i+j} \rangle.$$

В силу свойства мультипликативности

$$\langle \cos \theta_{i,i+j} \rangle = \cos^j \theta.$$

Проводим суммирование:

```

(%i10) assume(COS_THETA>0,COS_THETA<1)$
(%i11) sum(COS_THETA^j,j,1,N-i),simpsum$
(%i12) sum_theta:sum(%i,i,1,N),simpsum,factor;
(%o12) (COS_THETA * (COS_THETA^N - N - COS_THETA^N + 1)) / (COS_THETA - 1)^2

```

Заметим, что в случае  $\theta = \pi/2$  и  $\cos \theta = 0$  получается тот же результат, что и для свободно-сочленённой цепи. Там нулю был равен средний косинус, здесь — постоянный. Поэтому составим отношение  $\langle R^2 \rangle_{\theta} / \langle R^2 \rangle_{free}$  и рассмотрим предельные случаи. Для предельно короткой цепи  $N = 1$  и

```
(%i13) (N*a^2+2*a^2*%) / (N*a^2) $
(%i14) subst([N=1], %);
(%o14) 1
```

Для очень длинной цепи  $N \gg 1$ , поэтому применим предел  $N \rightarrow \infty$ .

```
(%i15) limit(%th(2), N, inf);
(%o15)  $\frac{\cos \theta + 1}{\cos \theta - 1}$ 
```

То есть для длинных цепей

$$\langle R^2 \rangle_\theta = Na^2 \frac{1 + \cos \theta}{1 - \cos \theta}$$

Найдём разность между полным отношением и его предельной частью:

```
(%i16) 1+2*sum_theta/N-%, factor;
(%o16)  $\frac{2 \cos \theta (\cos \theta^N - 1)}{(\cos \theta - 1)^2 N}$ 
```

и перепишем в виде:

$$\langle R^2 \rangle_\theta = Na^2 \left[ \frac{1 + \cos \theta}{1 - \cos \theta} - \frac{2}{N} \cos \theta \frac{1 - \cos^N \theta}{(1 - \cos \theta)^2} \right]$$

#### 1.4. Цепь с заторможенным внутренним вращением

Здесь, наряду с фиксацией валентного угла, существует и отличный от нуля потенциал вращения вокруг ординарных валентных связей  $U(\varphi)$ , в силу которого значения торсионного угла  $\varphi$  перестают быть равновероятными. Теперь их вероятность описывается больцмановской экспонентой

$$P(\varphi) \sim e^{-U(\varphi)/kT},$$

где  $k$  — постоянная Больцмана,  $T$  — температура. Отличен от нуля и средний косинус торсионного угла

$$\langle \cos \varphi \rangle = \frac{\int_0^{2\pi} \cos \varphi P(\varphi) d\varphi}{\int_0^{2\pi} P(\varphi) d\varphi}$$

Выкладки, которые мы здесь не рассматриваем, дают для больших  $N$

$$\langle R^2 \rangle_{\theta, \varphi} = Na^2 \frac{1 + \cos \theta}{1 - \cos \theta} \cdot \frac{1 + \langle \cos \varphi \rangle}{1 - \langle \cos \varphi \rangle}$$

#### 1.5. Сегмент Куна

Как видим, для всех трёх рассмотренных моделей средний квадрат расстояния между концами цепи имеет общий вид произведения

$$\langle R^2 \rangle = Na^2 \Theta(\theta) \Phi(\varphi),$$

причём при переходе от более сложных моделей к более простым два последних сомножителя последовательно становятся равными единице (вырождаются). Следовательно, цепи с заторможенным внутренним

вращением можно сопоставить свободно-сочленённую цепь с тем же  $\langle \mathbf{R}^2 \rangle$ , состоящую из так называемых сегментов Куна длиной  $A$  числом  $n$ :

$$Na^2\theta(\theta)\Phi(\varphi) = nA^2.$$

Чтобы получить второе уравнение для нахождения двух неизвестных ( $A$  и  $n$ ), введём также равенство контурных длин двух цепей, учитывая, что сохранение валентного угла приводит к пилообразной геометрии полностью вытянутой цепи:

$$Na \cos \frac{\theta}{2} = nA.$$

Поскольку  $\theta$  присутствует в обоих уравнениях, выпишем зависимости от него явно, функцию же  $\Phi(\varphi)$  расписывать не будем.

```
(%i17) n*A^2=N*a^2*(1+cos(theta))/(1-cos(theta))*Phi$
(%i18) n*A=N*a*cos(theta/2)$
(%i19) solve([%,%th(2)],[A,n]),factor;

(%o19) [ [ A = -\frac{a \Phi(\cos(\theta)+1)}{\cos(\frac{\theta}{2})(\cos(\theta)-1)}, n = -\frac{\cos(\frac{\theta}{2})^2(\cos(\theta)-1)N}{\Phi(\cos(\theta)+1)} ] ]

(%i20) first(first(%)),trigreduce,factor;

(%o20) A = -\frac{a \Phi \sec(\frac{\theta}{2})(\cos(\theta)+1)}{\cos(\theta)-1}
```

Замена реальной цепи на цепь куновских сегментов очень удобна, поскольку выражения и, значит, выкладки для свободно-сочленённой цепи намного проще (см. напр. следующую главу).

Заметим, что число сегментов Куна  $n$  может и не быть целым.

## 1.6. Предельный переход

В рассматривавшихся выше моделях гибкость цепи обусловлена шарнирными сочленениями между жёсткими прямыми сегментами. Она максимальна в точках смыкания сегментов и нулевая по всей остальной длине макромолекулы.

Вследствие этого можно представить себе противоположную модель, в которой названная гибкость равномерно рассредоточена (как бы «размазана») по всей длине макромолекулы, которую можно трактовать как тонкую упругую однородную нить, деформирующуюся по закону Гука (червеобразная, или персистентная модель). Математически переход между моделями достигается одновременным устремлением  $a \rightarrow 0$ ,  $N \rightarrow \infty$ ,  $\theta \rightarrow 0$ , причём должна сохраняться контурная длина цепи  $L = Na$ , а также появиться

некая новая величина, характеризующая степень гибкости новой модели макромолекулы.

Выполним упражнение из п. 3.4 [1, с. 27] (переход между моделями).

В силу малости переменных  $a$  и  $\theta$ , по ним в ряд Тейлора (Маклорена) можно разлагать те или иные выражения, ограничиваясь одним-двумя членами. Так, разложим  $\cos \theta$  в ряд, оставим два члена и подставим в формулу для  $\langle R^2 \rangle_\theta$ .

```
(%i21) taylor(cos(theta),theta,0,2);
(%o21) /T/ 1 -  $\frac{\theta^2}{2}$  + ...
(%i22) pickpart(% ,1);
(%t22)  $-\frac{\theta^2}{2}$ 
(%o22) 1 + %t22 + ...
(%i23) subst([COS_THETA=1+%t22],N*a^2+2*a^2*sum_theta),expand;
(%o23)  $\frac{4 a^2 N}{\theta^2} - a^2 N + \frac{8 a^2 \left(1 - \frac{\theta^2}{2}\right)^{N+1}}{\theta^4} + \frac{4 a^2}{\theta^2} - \frac{8 a^2}{\theta^4}$ 
```

Теперь нужно обработать наиболее сложный член, где  $N$  стоит в показателе степени: выделим его, возьмём логарифм, разложим в ряд Тейлора (оставив один член) и пропотенцируем, после чего вернём в общее выражение.

```

(%i24) pickapart(% , 3);
(%t24) a2
(%t25)  $\left(1 - \frac{\theta^2}{2}\right)^{N+1}$ 
(%t26)  $\theta^4$ 
(%o26)  $\frac{4 a^2 N}{\theta^2} - a^2 N + \frac{4 a^2}{\theta^2} + \frac{8 a^2}{\theta^4} - \frac{8 a^2}{\theta^6}$ 
(%i27) log(%t25), logexpand=super;
(%o27)  $\log\left(1 - \frac{\theta^2}{2}\right)(N+1)$ 
(%i28) taylor(% , theta, 0, 2);
(%o28)  $T/ - \frac{(N+1)\theta^2}{2} + \dots$ 
(%i29) pickapart(% , 1);
(%t29)  $\frac{(N+1)\theta^2}{2}$ 
(%o29) %t29
(%i30) exp(%t29);
(%o30)  $e^{\frac{\theta^2(N+1)}{2}}$ 
Копируем %o26 и заменяем %t25
(%i31)  $(4*a^2*N)/\theta^2 - a^2*N + (4*a^2)/\theta^2 + (8*a^2)/\theta^4 - (8*a^2)/\theta^6$ 
(%o31)  $\frac{8 a^2 e^{\frac{\theta^2(N+1)}{2}}}{\theta^4} + \frac{4 a^2 N}{\theta^2} - a^2 N + \frac{4 a^2}{\theta^2} - \frac{8 a^2}{\theta^4}$ 

```

Поскольку  $N \gg 1$ , то можно пренебречь слагаемым  $4a^2/\theta^2$  по сравнению с  $4Na^2/\theta^2$ .

```
(%i32) pickpart(% , 1);
```

$$(\%t32) \frac{8 a^2 e^{\frac{\theta^2 (N+1)}{2}}}{\theta^4}$$

$$(\%t33) \frac{4 a^2 N}{\theta^2}$$

$$(\%t34) -a^2 N$$

$$(\%t35) \frac{4 a^2}{\theta^2}$$

$$(\%t36) -\frac{8 a^2}{\theta^4}$$

```
(%o36) %t36 + %t35 + %t34 + %t33 + %t32
```

Копируем и убираем %t35

```
(%i37) %t36+%t34+%t33+%t32;
```

$$(\%o37) \frac{8 a^2 e^{\frac{\theta^2 (N+1)}{2}}}{\theta^4} + \frac{4 a^2 N}{\theta^2} - a^2 N - \frac{8 a^2}{\theta^4}$$

Как уже говорилось, произведение  $L = Na$  (контурная длина цепи) должно остаться конечным. Поскольку в Maxima трудно провести замену выражения на переменную в силу того, что части этого выражения могут не находиться рядом во внутреннем представлении, применим особый приём. Формально рассматривая  $L = Na$  как уравнение, решаем его относительно  $N$  (но не  $a$ , которое частично ещё остаётся в выражении) и подставляем в преобразуемое выражение.

```
(%i38) L=N*a$ solve(% , N);
```

$$(\%o39) [N = \frac{L}{a}]$$

```
(%i40) subst(% , %th(3));
```

$$(\%o40) \frac{8 a^2 e^{\theta^2 \left(\frac{L}{a} + 1\right)}}{\theta^4} + \frac{4 a L}{\theta^2} - a L - \frac{8 a^2}{\theta^4}$$

Замечаем, что две устремляемые к нулю величины,  $a$  и  $\theta$ , стоят в выражении в устойчивой комбинации  $a/\theta^2$ , имеющей размерность длины.

Учитывая также множитель 2, чтобы упростить аргумент экспоненты, вводим персистентную длину<sup>2</sup>  $l_p$ :

$$l_p \equiv \frac{2a}{\theta^2}.$$

Для внедрения её в выражение пользуемся только что описанным приёмом.

```
(%i41) lp=2*a/theta^2$ solve(% , theta);
(%o42) [theta=-sqrt(2)*sqrt(a/lp), theta=sqrt(2)*sqrt(a/lp)]
(%i43) subst(second(%), %th(3));
          a(L/a+1)
(%o43) 2 lp^2 %e^-lp + 2 lp L - a L - 2 lp^2
```

Оставшиеся слагаемые с  $a$  далее обнуляются. Кроме того, следуя автору формулы Гюнтеру Порода, представим конечное выражение как произведение сомножителя  $l_p^2$ , определяющего размерность квадрата длины, на функцию от безразмерного отношения  $L/l_p$ .

```
(%i44) limit(% , a, 0), expand;
(%o44) 2 lp^2 %e^-lp + 2 lp L - 2 lp^2
(%i45) % / (2*lp^2), expand;
(%o45) %e^-lp + L/lp - 1
```

Таким образом, при предельном переходе  $\{a \rightarrow 0, N \rightarrow \infty, \theta \rightarrow 0\}$  формула для среднего квадрата расстояния между концами цепи из трёхпараметрической  $\{N, a, \theta\}$  становится двухпараметрической  $\{L, l_p\}$  и приобретает вид

$$\langle R^2 \rangle_p = 2l_p^2 \left( \frac{L}{l_p} - 1 + e^{-L/l_p} \right).$$

С другим видом формулы Порода мы познакомимся в главе 3.

<sup>2</sup> Обычно эту величину обозначают буквой  $l$  с тильдой сверху. Так как в Maxima нет диакритических знаков, мы сочли целесообразным заменить тильду нижним индексом.

## 2. Функции распределения в моделях идеальной полимерной цепи

В предыдущей главе мы рассмотрели средние величины, характеризующие различные варианты модели идеальной полимерной цепи. Более подробно такая модель описывается функцией распределения вектора между её концами  $P_N(\mathbf{R})$ , то есть плотностью вероятности того, что вектор, соединяющий начало и конец  $N$ -звенной цепи, равен  $\mathbf{R}$ . Это коэффициент пропорциональности в выражении

$$dW(\mathbf{R}) = P_N(\mathbf{R})d\mathbf{R},$$

причём  $d\mathbf{R} = dx dy dz$ .

Подробно вывод этой функции дан в учебнике [3], к которому и отсылаем читателя за деталями. Здесь мы даём выкладки в сокращении.

### 2.1. Задача складного метра

Задача о блуждании второго конца цепи в пространстве, когда первый приурочен к началу координат, сводится к задаче блуждания его проекций на три равноправные оси —  $Ox$ ,  $Oy$  и  $Oz$ . Поэтому сначала рассмотрим одномерный вариант нашей задачи, известный как *задача складного метра*.

Пусть складной метр состоит из  $N$  шарнирно соединённых сегментов длиной  $b$ . Начало координат  $x = 0$  приурочено к началу первого сегмента. Метр раскладывается случайным образом, так что вероятность любого его сегмента располагаться слева направо или справа налево одинакова и равна  $1/2$ . Найти вероятность  $P(x)$  того, что конец последнего сегмента попадёт в некую точку с координатой  $x$ .

Заметим, что  $x$  должно быть кратно  $b$ , а распределение вероятностей — дискретным. Однако затем мы перейдём к непрерывному распределению вероятностей, используя тот факт, что для полимерных цепей  $N \gg 1$ .

Вероятность в схеме случаев определяется как отношение числа случаев, благоприятствующих данному событию, к общему числу всех возможных случаев. Здесь общее число случаев (вариантов раскладки метра) равно  $2^N$ . Пусть  $N_1$  сегмент расположен слева направо, а  $N_2$  — справа налево. Тогда  $N_1 + N_2 = N$ ,  $(N_1 - N_2)b = x$ . И наоборот, если задана координата  $x$ , то можно однозначно найти число тех и других сегментов.

```
(%i1) assume (N>0, a>0, b>0, A>0, beta>0) $
(%i2) N1+N2=N$ (N1-N2)*b=x$
(%i4) solve([%, %th(2)], [N1, N2]), expand;
(%o4) [ [ N1 =  $\frac{N}{2} + \frac{x}{2b}$ , N2 =  $\frac{N}{2} - \frac{x}{2b}$  ] ]
```

Следовательно,  $N_1 = (N + x/b)/2$ ,  $N_2 = (N - x/b)/2$ .

Согласно комбинаторике, число перестановок  $N$  объектов, которым соответствуют числа сегментов  $N_1$  и  $N_2$ , равно

$$\frac{N!}{N_1!N_2!}$$

Относя это число к общему числу всех вариантов, получим вероятность реализации расстояния  $x$ :

$$P(x) = \frac{N!}{2^N N_1! N_2!}$$

С факториалами неудобно иметь дело, тем более что при больших аргументах ( $N, N_1, N_2$ ) можно с хорошим приближением применить формулу Стирлинга. При этом учтём, что нормировка на единицу (т.е. сумма всех вероятностей должна равняться единице) при упрощениях не сохраняется, поэтому в конце выкладок придётся произвести перенормировку. А раз так, то компоненты выражения для  $P$ , не зависящие от  $x$ , можно опустить: это  $N!$  в числителе и  $2^N$  в знаменателе. Запишем знаменатель, помня, что  $n! = \Gamma(n+1)$ , и применим формулу Стирлинга:

```
(%i5) load (stirling)$
(%i6) gamma(rhs(first(first(%th(2))))+1)*gamma(rhs(second(first(%th(2))))+1);
(%o6)  $\Gamma\left(\frac{N}{2} - \frac{x}{2b} + 1\right) \Gamma\left(\frac{N}{2} + \frac{x}{2b} + 1\right)$ 
(%i7) stirling(% , 1);
(%o7)  $2 \pi \left(\frac{N}{2} - \frac{x}{2b} + 1\right)^{N/2 - \frac{x}{2b} + \frac{1}{2}} \left(\frac{N}{2} + \frac{x}{2b} + 1\right)^{N/2 + \frac{x}{2b} + \frac{1}{2}} e^{-N-2}$ 
```

Возьмём логарифм:

```
(%i8) log(%), logexpand=super;
(%o8)  $-N + \log\left(\frac{N}{2} + \frac{x}{2b} + 1\right) \left(\frac{N}{2} + \frac{x}{2b} + 1\right) + \log\left(\frac{N}{2} - \frac{x}{2b} + 1\right) \left(\frac{N}{2} - \frac{x}{2b} + 1\right) + \log(\pi) + \log(2) - 2$ 
```

Ввиду будущей перенормировки проигнорируем слагаемые, не зависящие от  $x$ , для чего «разберём» преобразуемое выражение:

```
(%i9) pickpart(% , 1);
(%t9)  $\log\left(\frac{N}{2} + \frac{x}{2b} + 1\right) \left(\frac{N}{2} + \frac{x}{2b} + 1\right)$ 
(%t10)  $\log\left(\frac{N}{2} - \frac{x}{2b} + 1\right) \left(\frac{N}{2} - \frac{x}{2b} + 1\right)$ 
(%t11) log(pi)
(%t12) log(2)
(%o12)  $-N + \%t9 + \%t12 + \%t11 + \%t10 - 2$ 
(%i13) %t9+%t10;
(%o13)  $\log\left(\frac{N}{2} + \frac{x}{2b} + 1\right) \left(\frac{N}{2} + \frac{x}{2b} + 1\right) + \log\left(\frac{N}{2} - \frac{x}{2b} + 1\right) \left(\frac{N}{2} - \frac{x}{2b} + 1\right)$ 
```

Поскольку  $N \gg 1$ , вручную убираем несравнимые с ним  $\frac{1}{2}$  и 1 из скобок:

```
(%i14) log(N/2+x/(2*b)) * (N/2+x/(2*b)) + log(N/2-x/(2*b)) * (N/2-x/(2*b));
(%o14) log((N+x)/(2*b)) * ((N+x)/(2*b)) + log((N-x)/(2*b)) * ((N-x)/(2*b))
```

Раскладываем в ряд Тейлора и снова игнорируем слагаемые без  $x$ :

```
(%i15) taylor(%, x, 0, 3);
(%o15) T/ (-log(2)+log(N)) N +  $\frac{x^2}{2 b^2 N} + \dots$ 
(%i16) pickpart(%, 1);
(%t16) (-log(2)+log(N)) N
(%t17)  $\frac{x^2}{2 b^2 N}$ 
(%o17) %t16 + %t17 + ...
```

Напомним, что мы преобразовывали логарифм знаменателя вероятности. Возвратимся к самой вероятности, которая пока получается ненормированной:

```
(%i18) exp(-%t17);
(%o18)  $\%e^{-\frac{x^2}{2 b^2 N}}$ 
```

На этом этапе уже можно считать полученное выражение *непрерывной* функцией распределения *действительной* величины  $x$ , поэтому нормировка выражается уравнением:

$$\int_{-\infty}^{\infty} C e^{-x^2/2Nb^2} dx = 1,$$

откуда и находим нормировочный делитель. Делим на него:

```
(%i19) integrate(%, x, -inf, inf);
(%o19)  $\sqrt{2} \sqrt{\pi} b \sqrt{N}$ 
(%i20) %th(2)/%;
(%o20)  $\frac{\%e^{-\frac{x^2}{2 b^2 N}}}{\sqrt{2} \sqrt{\pi} b \sqrt{N}}$ 
```

## 2.2. Функция распределения вектора между концами цепи

Чтобы полученным в предыдущем разделе результатом можно было воспользоваться для нахождения искомого трёхмерного распределения  $P_N(\mathbf{R})$ , необходимо, чтобы попадание координаты  $x$  конца цепи в диапазон  $[x, x+dx]$ , координаты  $y$  в диапазон  $[y, y+dy]$ , координаты  $z$  в диапазон  $[z, z+dz]$  были *независимыми* событиями, чьи вероятности перемножаются согласно

теореме умножения вероятностей. А для этого движение сегментов цепи должно быть некоррелированным, что имеет место лишь для сегментов Куна при мысленной замене реальной цепи на свободно-сочленённую (см. § 1.5).

Перемножаем вероятности для трёх осей координат:

```
(%i21) %*subst([x=y],%) *subst([x=z],%), radcan;
```

$$\frac{e^{-\frac{z^2+y^2+x^2}{2b^2N}}}{2^{3/2} \pi^{3/2} b^3 N^{3/2}}$$

```
(%o21)
```

Конечно, при блуждании проекций сегментов по осям длины этих проекций не будут оставаться неизменными, поэтому вместо однозначной величины  $b$  следует подставить некую среднюю величину. Знаменатель получившегося выражения можно представить как  $(2\pi N b^2)^{3/2}$ , в аргументе экспоненты тоже стоит  $b^2$ , откуда видно, что  $b^2$  здесь можно придать смысл среднего квадрата проекции сегмента Куна на любую ось.

Мгновенная проекция сегмента на любую ось равна  $b = A \cos \theta$ , где  $A$  — длина сегмента Куна,  $\theta$  — угол, который он составляет с выбранной осью. Соответственно,  $\langle b^2 \rangle = A^2 \langle \cos^2 \theta \rangle$ .

В силу свойств свободного сочленения все возможные значения  $\theta$  равновероятны, откуда

$$\langle \cos^2 \theta \rangle = \frac{\int_0^\pi \cos^2 \theta \sin \theta d\theta}{\int_0^\pi \sin \theta d\theta}.$$

```
(%i22) A*cos(theta)$
```

```
(%i23) integrate(%^2*sin(theta),theta,0,%pi) /
```

```
integrate(sin(theta),theta,0,%pi);
```

```
(%o23) A^2 / 3
```

Итак,  $\langle b^2 \rangle = A^2/3$ . Тогда  $N$  — это число сегментов Куна, которое раньше обозначалось как  $n$  (см. § 1.5).

```
(%i24) subst([b=sqrt(%),N=n],%th(3));
```

$$\frac{3(z^2+y^2+x^2)}{2nA^2}$$

```
(%o24) \frac{3^{3/2} e^{-\frac{3(z^2+y^2+x^2)}{2nA^2}}}{2^{3/2} \pi^{3/2} n^{3/2} A^3}
```

Видим, что в получившемся выражении для трёхмерной функции распределения координаты конца вектора входят в комбинации  $x^2+y^2+z^2$ , в которой узнаваем квадрат длины этого вектора  $\mathbf{R}^2$ . Проводим замену:

```
(%i25) subst([x^2+y^2+z^2=R^2],%);
          3 R^2
          -----
(%o25) 3^{3/2} e^{-2 n A^2}
      2^{3/2} pi^{3/2} n^{3/2} A^3
```

Что же касается параметров распределения  $n$  и  $A$ , то они входят в выражение дважды в одной и той же комбинации  $nA^2$ .

Вопрос: если эти параметры входят в аргумент экспоненты в комбинации  $nA^2$ , в какой иной комбинации они могут входить в оставшуюся часть выражения, представляющую собой нормировочный множитель для этой экспоненты?

Проводим замену:

```
(%i26) 3/(2*n*A^2)=beta^2$
(%i27) solve(% ,n);
(%o27) [n=3/(2*beta^2*A^2)]
(%i28) subst(% ,%th(3));
(%o28) beta^3 e^{-beta^2 R^2}
      pi^{3/2}
```

Таким образом,

$$P_N(\mathbf{R}) = \left(\frac{\beta}{\sqrt{\pi}}\right)^3 e^{-\beta^2 \mathbf{R}^2}.$$

Это распределение Гаусса. Отличительные черты: экспонента и минус квадрат аргумента под ней, больше аргумент нигде не фигурирует.

**2.3. Энтропийная природа эластичности одиночной цепи**

Полученная выше функция, в силу определения вероятности, пропорциональна числу вариантов, которыми реализуется заданное значение  $\mathbf{R}^2$ . Максимально это число при  $\mathbf{R}^2 = 0$ , то есть когда конец цепи совпадает с её началом, это наиболее вероятное состояние. А поскольку, согласно формуле Больцмана, энтропия равна

$$S = k \ln W,$$

где  $W$  — число микросостояний, то состояние с  $\mathbf{R} = 0$  отвечает и максимальной энтропии цепи, рассматриваемой как термодинамическая система.

С отклонением  $\mathbf{R}$  от нулевого и вероятность состояния, и энтропия цепи убывают, а свободная энергия Гельмгольца  $F = U - TS$  возрастает (внутренняя энергия идеальной цепи  $U$  полагается постоянной, так что  $\Delta U \approx 0$ ), то есть для растяжения цепи над ней нужно совершить работу внешних

сил  $\Delta F = \Delta U - T\Delta S$ . Значит, одиночная полимерная цепь обладает свойством эластичности, сопротивления растягивающему усилию. Найдём числовую характеристику этого свойства.

$$\Delta S(R) = S(R) - S(0) = k \ln \frac{W_R}{W_0}, \quad \Delta F \approx -T\Delta S, \quad \mathbf{f} = \frac{\partial \Delta F}{\partial \mathbf{R}},$$

где  $\mathbf{f}$  — сила растяжения на вектор  $\mathbf{R}$ . Считаем:

```
(%i29) S:k*log(%)$
(%i30) diff(-%*T,R);
(%o30) 2 beta^2 k R T
```

Таким образом,  $\mathbf{f} = 2kT \beta^2 \mathbf{R}$ , то есть выполняется закон Гука: сила пропорциональна деформации (хотя и не относительной). Можно раскрыть  $\beta$ :

```
(%i31) subst([beta^2=3/(2*n*A^2)],%);
(%o31) 3 k R T
      n A^2
```

#### 2.4. Функция распределения расстояния между концами цепи

В выражение распределения для вектора  $\mathbf{R}$  он входит лишь в виде квадрата своей длины  $\mathbf{R}^2$ , то есть фактически это — скалярное распределение. Что неудивительно, если учесть анизотропию пространства — все направления в нём равноценны. Единственной векторной компонентой в распределении является элемент объёма  $dx dy dz$ , приуроченный к конкретной точке пространства. Перейдём к сферическим координатам, в которых элементарный объём равен  $dx dy dz = R^2 dR \sin \theta d\theta d\phi$ , что позволит учесть анизотропию пространства (равноценность всех углов  $\theta$  и  $\phi$ ) посредством интегрирования по этим углам:

```
(%i32) %th(4)*R^2*sin(theta);
(%o32) beta^3 sin(theta) R^2 %e^-beta^2 R^2
      pi^3/2
(%i33) integrate(% ,theta,0,%pi);
(%o33) 2 beta^3 R^2 %e^-beta^2 R^2
      pi^3/2
(%i34) PN:integrate(% ,phi,0,2*%pi);
(%o34) 4 beta^3 R^2 %e^-beta^2 R^2
      sqrt(pi)
```

Таким образом,

$$dW\{R, R + dR\} = \frac{4\beta^3}{\sqrt{\pi}} R^2 e^{-\beta^2 R^2} dR,$$

где  $R$  — длина отрезка между концами цепи, величина скалярная. В полученном выражении нетрудно узнать распределение Максвелла.

## 2.5. Свойства распределения расстояния между концами цепи

Полученная функция  $P_N(R)$  имеет один аргумент (скаляр  $R$ ) и один параметр ( $\beta$ ).

### 2.5.1. Положение максимума (наиболее вероятное расстояние между концами цепи)

Находим, как обычно, дифференцированием:

```
(%i35) diff(PN,R);
(%o35) 
$$\frac{8\beta^3 R e^{-\beta^2 R^2}}{\sqrt{\pi}} - \frac{8\beta^5 R^3 e^{-\beta^2 R^2}}{\sqrt{\pi}}$$

(%i36) solve(%,R);
(%o36) 
$$\left[ R = -\frac{1}{\beta}, R = \frac{1}{\beta}, R = 0 \right]$$

```

Выбираем, конечно, второй, положительный корень.

### 2.5.2. Среднее расстояние между концами цепи

Для скалярного распределения, в отличие от векторного, это величина ненулевая, первый момент распределения.

```
(%i37) integrate(R*PN,R,0,inf);
(%o37) 
$$\frac{2}{\sqrt{\pi}\beta}$$

```

### 2.5.3. Средний квадрат расстояния между концами цепи

Это — второй момент распределения:


```
(%i38) integrate(R^2*PN,R,0,inf);
(%o38) 
$$\frac{3}{2\beta^2}$$

```

Итак, мы получили две величины с размерностью длины и одну — с размерностью квадрата длины, каждую из которых можно взять для замены  $\beta$  в функции распределения. Выбираем последнюю, потому что: 1) она имеет размерность квадрата длины, а под экспонентой и вне её стоит  $R^2$ ; 2)  $1/\beta^2 \sim nA^2$ , а последняя величина, как мы установили в предыдущей главе (§ 1.5), инвариантна при переходе между моделями идеальной полимерной цепи.

```
(%i39) solve(%=R2,beta);
(%o39) [ $\beta = -\frac{\sqrt{3}}{\sqrt{2}\sqrt{R2}}$ ,  $\beta = \frac{\sqrt{3}}{\sqrt{2}\sqrt{R2}}$ ]
(%i40) subst(second(%),PN);
(%o40)  $\frac{\sqrt{2} 3^{3/2} R^2 e^{-\frac{3 R^2}{2 R2}}}{\sqrt{\pi} R2^{3/2}}$ 
```

Теперь можно построить график скалярного распределения. Если принять  $R2 = 1$ , то по оси абсцисс будет фактически отложено отношение  $R/\langle R^2 \rangle^{1/2}$ .



#### 2.5.4. Средняя флуктуация квадрата расстояния между концами цепи

В процессе броуновского движения расстояние между концами цепи флуктуирует, то же происходит и с его квадратом (по сути, с размером клубка). Средняя относительная флуктуация квадрата расстояния между концами цепи выражается формулой:

$$\frac{\langle (R^2 - \langle R^2 \rangle)^2 \rangle}{\langle R^2 \rangle^2} = \frac{\langle R^4 \rangle}{\langle R^2 \rangle^2} - 1.$$

Рассчитаем её по правой части:

```
(%i42) integrate (R^2*PN,R,0,inf);
(%o42) 3
      2 beta^2
(%i43) integrate (R^4*PN,R,0,inf);
(%o43) 15
      4 beta^4
(%i44) %/%th(2)^2-1;
(%o44) 2
      3
```

а расчёт по левой части оставим в качестве упражнения читателю.

Видим, что флуктуации  $\mathbf{R}^2$  — порядка среднего значения этой величины. Следовательно, гауссов клубок — сильно флуктуирующая система.

Упражнение. Найдите по аналогии флуктуацию первой степени  $\mathbf{R}$ .

## 2.6. Гидродинамический радиус клубка

Эта величина определяется формулой

$$R_D^{-1} = \frac{1}{2} N^{-2} \left\langle \sum_{i=1}^N \sum_{j=1, (i \neq j)}^N |\mathbf{r}_{ij}|^{-1} \right\rangle.$$

где  $\mathbf{r}_{ij}$  — вектор, соединяющий звенья  $i$  и  $j$ . Контурная длина этого отрезка цепи равна  $(j - i)a$ . Рассчитаем обратный радиус всей цепи и подставим  $(i - j)a^2$  вместо  $R^2$ :

```
(%i45) assume (i>2,R2>0) $
(%i46) integrate (%o40/R,R,0,inf);
(%o46) sqrt(2)*sqrt(3)
      sqrt(pi)*sqrt(R2)
(%i47) subst ([R2=(i-j)*a^2],%);
(%o47) sqrt(2)*sqrt(3)
      sqrt(pi)*a*sqrt(i-j)
```

Ср. сур. (5.4) из [1]:

$$\langle |\mathbf{r}_{ij}|^{-1} \rangle = (6/\pi)^{1/2} |i - j|^{-1/2} a^{-1}.$$

Подставляем в предыдущую формулу и заменяем суммы интегралами, что допустимо для  $N \gg 1$  (см. § 1.2).

```

(%i48) integrate(%, j, 1, i-1) $
(%i49) integrate(%, i, 1, N) $
(%i50) N^2/%;
(%o50) 
$$\frac{\sqrt{3}\sqrt{\pi} a N^2}{\sqrt{2}(\sqrt{N-1}(4N-4)-6N+6)}$$

Вручную копируем и убираем пренебрежимо малое
(%i51) (sqrt(3)*sqrt(%pi)*a*N^2)/(sqrt(2)*(sqrt(N)*(4*N)));
(%o51) 
$$\frac{\sqrt{3}\sqrt{\pi} a \sqrt{N}}{2^{5/2}}$$


```

Сравните с ур. (5.6) из [1]:

$$R_D = (3\pi/32)^{1/2} \langle R^2 \rangle^{1/2}.$$

### 2.7. Случай кольцевой макромолекулы

Представьте себе, что начало и конец свободно-сочленённой цепи соединены ковалентной связью и возникла кольцевая топология макромолекулы. Для такой цепи можно ставить вопрос о распределении вектора  $\mathbf{R}_{mn}$  между двумя произвольно выбранными сегментами  $m$  и  $n$ . Можно ли найти это распределение на основе уже известного для открытой линейной цепи?

Среди великого множества конформаций открытой цепи найдётся подмножество таких конформаций, начало и конец цепи в которых *случайно* совпадут. Это подмножество и будет полным набором конформаций кольцевой макромолекулы, и очевидно, что ковалентная связь между началом и концом ничего нового в этот набор внести не сможет. А среди подмножества, в котором  $\mathbf{R} = 0$ , найдётся своё подмножество, в котором вектор между сегментами  $m$  и  $n$  равен заданному  $\mathbf{R}_{mn}$ . Поэтому интересующая нас вероятность  $P_N^\circ(\mathbf{R}_{mn})$  есть *условная* вероятность того, что в *линейной* цепи из  $N$  звеньев вектор между звеньями  $m$  и  $n$  есть  $\mathbf{R}_{mn}$  при условии, что вектор между концами цепи нулевой:

$$P_N^\circ(\mathbf{R}_{mn}) = P_N(\mathbf{R}_{mn} | \mathbf{R} = 0).$$

По определению условной вероятности,

$$P_N(\mathbf{R}_{mn} | \mathbf{R} = 0) = \frac{P_N(\mathbf{R}_{mn}, \mathbf{R} = 0)}{P_N(\mathbf{R} = 0)},$$

где в числителе стоит совместная вероятность — выполняться должны оба условия одновременно. Знаменатель же служит нормирующим делителем, в него не входит вектор, распределение по которому ищется.

Вопрос: нельзя ли проигнорировать знаменатель и в дальнейшем перенормировать вероятность, как это делали в § 2.1?

Участки цепи, соединяющие сегменты  $m$  и  $n$  по и против часовой стрелки, независимы друг от друга, потому, согласно теореме умножения вероятностей,

$$P_N(\mathbf{R}_{mn}, \mathbf{R} = 0) = P_{n-m}(\mathbf{R}_{mn})P_{N-n+m}(\mathbf{R}_{mn}).$$

Представьте себе, что цепь разомкнута на сегменте  $n$ , тогда сложное событие будет состоять в том, чтобы оба «болтающихся» конца цепи одновременно попали в точку конца вектора  $\mathbf{R}_{mn}$ .

Считаем:

```
(%i52) forget(i>2)$ assume(i>0)$
(%i54) exp(-3*R^2/(2*a^2*i))$
(%i55) integrate(%,R,-inf,inf)$
(%i56) %th(2)/%;
```

$$\frac{\sqrt{3} e^{-\frac{3R^2}{2a^2i}}}{\sqrt{2}\sqrt{\pi}a\sqrt{i}}$$

```
(%i57) subst([i=n-m],%) * subst([i=N-n+m],%) /
subst([i=N,R=0],%) ;
```

$$\frac{\sqrt{3}\sqrt{N} e^{-\frac{3R^2}{2a^2(N-n+m)}}}{\sqrt{2}\sqrt{\pi}a\sqrt{n-m}\sqrt{N-n+m}}$$

Судя по  $(-R^2)$  в показателе экспоненты, сохраняется гауссова зависимость. Вручную копируем этот показатель и убираем  $R^2$  и  $a^2$ :

```
(%i58) -1/(N-n+m)-1/(n-m),factor;
```

$$\frac{N}{(n-m)(N-n+m)}$$

Сравниваем с предэкспоненциальным (нормирующим) множителем. Вводим замену и преобразуем:

```
(%i59) mu=-1/;%$ solve(%,N);
```

$$\left[ N = \frac{n^2 - 2mn + m^2}{n - \mu - m} \right]$$

```
(%i61) subst(%,%th(4)),radcan;
```

$$\frac{\sqrt{3} e^{-\frac{3R^2}{2a^2\mu}}}{\sqrt{2}\sqrt{\pi}a\sqrt{\mu}}$$

Получили гауссово распределение, в котором вместо числа звеньев  $N$  стоит число  $\mu = (n - m)(N - n + m)/N$ .

Упражнение. Пренебрегите третьим subst в %i57 и после вывода нормируйте получившееся распределение. Сравните с результатом выше.

Используя полученное распределение, рассчитаем средний квадрат радиуса инерции (см. § 1.2) двумя способами:

```
(%i62) subst(%th(3), mu*a^2);
(%o62) 
$$\frac{a^2 (n-m)(N-n+m)}{N}$$

.
(%i63) sum(%, m, 1, n), simpsum$
(%i64) sum(%, n, 1, N), simpsum$
(%i65) %/N^2, expand;
(%o65) 
$$\frac{a^2 N}{12} - \frac{a^2}{12 N}$$

.
(%i66) integrate(%th(4), m, 1, n)$
(%i67) integrate(%, n, 1, N)$
(%i68) %/N^2, factor;
(%o68) 
$$\frac{a^2 (N-1)^3 (N+1)}{12 N^3}$$

```

См. § 1.2 для учёта условия  $N \gg 1$ .

Как видите, средний квадрат радиуса инерции уменьшился *вдвое* по сравнению с разомкнутой цепью.

Упражнение. Рассчитайте средний гидродинамический радиус кольцевой макромолекулы.

### 3. Модель червеобразной цепи. Персистентная длина

В разделе 1.6 было показано, что при рассредоточении гибкости вдоль всей макромолекулы последняя описывается лишь двумя параметрами — контурной длиной  $L$  и персистентной длиной  $l_p$ , характеризующей собственно гибкость. Подойдём теперь к этому вопросу с другой стороны.

#### 3.1. Корреляция между положениями сегментов

В разделе 1.3 фигурировал средний косинус между  $i$ -м и  $(i+k)$ -м сегментами цепи:

$$\langle \cos \vartheta_{i,i+k} \rangle = \cos^k \theta.$$

Эта величина количественно описывает корреляцию между направлениями этих сегментов в пространстве при (микро)броуновском движении. Когда это движение настолько свободное, что корреляция отсутствует,  $\langle \cos \vartheta_{i,i+k} \rangle = 0$ ; в другом предельном случае, случае абсолютной жёсткости цепи (отсутствие микроброуновского движения),  $\cos \theta = 1$  и  $\langle \cos \vartheta_{i,i+k} \rangle = 1$  при любом числе промежуточных сегментов  $k$ . В промежуточных случаях  $0 < \cos \theta < 1$ , и корреляция убывает с ростом числа промежуточных сегментов  $k$  по степенному закону.

Но число промежуточных сегментов пропорционально длине соответствующего участка цепи. Заменим  $k$  на  $s/a$ , где  $s$  — длина участка между звеньями  $i$  и  $(i+k)$ ,  $a$  — длина сегмента, и перейдём к экспоненциальной зависимости.

```
(%i1) assume(COS>0,COS<1,L>0,lp>0)$
(%i2) COS^k$ subst([k=s/a],%);
(%o3) COS^s/a
(%i4) exp(log(%));
          s log(COS)
(%o4) %e  a
```

Таким образом, можно записать:

$$\langle \cos \vartheta(s) \rangle = e^{\ln \cos \theta \cdot s/a}.$$

Заметим, что  $\cos \theta < 1$  и  $\ln \cos \theta < 0$ , то есть зависимость — обратная экспоненциальная, затухающая с расстоянием. При такой записи мы не ограничены значениями  $s$ , кратными  $a$ . Перепишем, явно указывая знак:

$$\frac{\ln \cos \theta \cdot s}{a} = -\frac{s}{(-a/\ln \cos \theta)} = -\frac{s}{l_p},$$

где  $l_p$  — персистентная длина.

```
(%i5) a/log(COS)=-lp$ solve(% ,a);
(%o6) [ a=-lp log(COS)]
(%i7) subst(% ,%th(3));
      s
(%o7) %e -lp
```

Теперь, когда два параметра цепи,  $a$  и  $\theta$ , редуцировались до одного,  $l_p$ , мы можем рассматривать цепь как непрерывную, червеобразную, без деления на отдельные сегменты. При  $s = l_p$  имеем:  $\langle \cos \vartheta \rangle = e^{-1}$ , поэтому персистентную длину можно определить как длину такого участка цепи, на котором средний корреляционный косинус равен  $e^{-1} \approx 0.368$ .

Чтобы установить тождество  $l_p$  из раздела 1.6 с введённой здесь персистентной длиной, заметим, что при малых  $\theta$  можно произвести разложение в ряд Тейлора с минимальным числом членов

```
(%i8) taylor(-a/log(cos(theta)), theta, 0, 1);
(%o8) /T/ 2 a a
      2 3 + ...
```

Сравните с %i41 из § 1.6, учитывая, что  $a \rightarrow 0$ .

### 3.2. Проекция цепи на направление первого сегмента

Найдём проекцию всей червеобразной цепи на направление её первого сегмента:

$$\langle X \rangle = \int_0^L \langle \cos \vartheta(s) \rangle ds = \int_0^L e^{-s/l_p} ds$$

и устремим длину цепи к бесконечности:

```
(%i9) integrate(%th(2), s, 0, L);
      L
(%o9) lp - lp %e -lp
(%i10) limit(% , L, inf);
(%o10) lp
```

Поскольку червеобразная цепь описывается двумя параметрами ( $L$ ,  $l_p$ ), из которых один ( $L$ ) исчезает, стремясь к бесконечности, можно предположить, что величина с размерностью длины (проекция) будет пропорциональна оставшемуся параметру ( $l_p$ ), который также имеет размерность длины. Видим, что коэффициент пропорциональности равен единице, поэтому названную проекцию можно считать альтернативным определением персистентной длины.

### 3.3. Средний квадрат длины цепи

Для непрерывной цепи вектор  $\mathbf{R}$ , соединяющий её начало с концом, запишется в виде уже не дискретной суммы, а интеграла

$$\mathbf{R} = \int_0^L \mathbf{u}(s) ds,$$

где  $\mathbf{u}(s)$  — единичный вектор направления цепи в точке, отстоящей на расстояние  $s$  вдоль контура цепи от её начала. Соответственно,  $\langle \mathbf{R}^2 \rangle$  вычисляется как двойной интеграл

$$\langle \mathbf{R}^2 \rangle = \int_0^L ds \int_0^L ds' \langle \mathbf{u}(s) \mathbf{u}(s') \rangle = 2 \int_0^L ds \int_0^{L-s} dt \langle \cos \theta(t) \rangle,$$

где  $t = s' - s$ .

```
(%i11) integrate(subst([s=t],%th(4)),t,0,L-s);
(%o11) lp - lp %es/lp - L/lp
(%i12) 2*integrate(%s,0,L);
(%o12) 2 (lp2 %e-L/lp + lp L - lp2)
```

Ср. с %o44 из § 1.6.

Для коротких цепей  $L \ll l_p$ , поэтому разложим  $\langle \mathbf{R}^2 \rangle$  в ряд Тейлора по  $L$  вокруг нуля и ограничимся членами второго порядка:

```
(%i13) taylor(%L,0,2);
(%o13)/T/ L2+...
```

Для длинных цепей, наоборот,  $l_p \ll L$ , поэтому разложение надо производить по  $l_p$ :

```
(%i14) taylor(%th(2),lp,0,1);
(%o14)/T/ 2 L lp+...
```

Таким образом (см. ур. 3.7 из [1]):

$$\langle \mathbf{R}^2 \rangle = \begin{cases} L^2, & L \ll l_p; \\ 2Ll_p, & L \gg l_p. \end{cases}$$

Здесь приведён упрощённый вариант расчёта. Более строго, следовало бы получить часть выражения относительно малой величины  $L/l_p$  (первый случай) или  $l_p/L$  (второй случай) и разлагать в ряд Тейлора только эту часть по этой малой переменной, а затем возвращаться к  $L$  и  $l_p$ . Оставим это читателю как упражнение.

### 3.4. Альтернативная форма уравнения Порода. Его численное решение


Выражение для  $\langle R^2 \rangle$  длинных цепей можно использовать для несколько иной записи формулы Порода. А именно, составим отношение  $\langle R^2 \rangle$  к  $2Ll_p$  (это  $\langle R^2 \rangle$  для длинных цепей):

```
(%i15) %th(3)/(2*lp*L),expand;
(%o15) 
$$\frac{lp e^{-\frac{L}{lp}} - lp}{L} + 1$$

(%i16) lambda=L/lp$ solve(%,L);
(%o17) [L=lp lambda]
(%i18) Rho2:subst(%,%th(3));
(%o18) 
$$\frac{e^{-lambda} - 1}{lambda} + 1$$

```

Получили функцию от отношения  $L/l_p$ , которое обозначили за  $\lambda$  — длина цепи в единицах персистентной длины. Построим график.



Как и ожидалось, кривая  $\langle R^2 \rangle / 2Ll_p (\lambda)$  выходит из нуля, монотонно растёт и в пределе длинных цепей асимптотически стремится к единице. Этот график наглядно показывает отличия цепи с конкретным конечным  $\lambda$  от предельно длинной цепи, как бы беря последнюю за эталон.

Уравнение Порода нельзя численно решить аналитически, то есть найти  $\lambda$ , зная  $\langle R^2 \rangle / 2Ll_p$ . Проверим:

```
(%i20) solve(Rho2=rho2,lambda);
(%o20) [lambda = -
$$\frac{e^{-lambda}(e^{lambda} - 1)}{rho2 - 1}$$
]
```

Решить уравнение можно только численно. В арсенале MAXIMA есть для этой цели стандартная функция `find_root`, реализующая метод

половинного деления (его описание можно найти в любом учебнике по численным методам). Для её работы нужно задать начальную «вилку»: какое-либо значение аргумента заведомо меньше искомого корня (левее на графике), и ещё одно — больше (правее).

Представьте себе, что на предыдущем графике проведена горизонтальная прямая с некоей ординатой  $\langle \mathbf{R}^2 \rangle / 2Ll_p$ , которую для краткости обозначим как  $\rho^2$ . Её пересечение с кривой даст искомый корень. Чтобы получить таким же образом значение  $\lambda$  левее этого корня, нужно, очевидно, график сместить вверх, не нарушая асимптотику при больших  $\lambda$ . Для этого уравнение следует соответствующим образом изменить, причём так, чтобы стало возможным его аналитическое решение, которое и даст необходимую левую границу.

Для получения правой границы кривую нужно сдвинуть соответственно вниз.

Перепишем уравнение Порода в виде:

$$\rho^2 = 1 - \frac{1 - 1/e^\lambda}{\lambda} = 1 - \frac{1}{\lambda} - \frac{1}{\lambda e^\lambda}.$$

Вспомним разложение экспоненты в ряд Тейлора:

```
(%i21) taylor(exp(x), x, 0, 3);
(%o21) /T/ 1 + x +  $\frac{x^2}{2}$  +  $\frac{x^3}{6}$  + ...
```


Если оборвать этот знакопостоянный ряд на втором члене и подставить вместо  $e^\lambda$  в знаменатель, то последний уменьшится, а всё выражение целиком — увеличится. Решая получившееся уравнение аналитически, находим левую границу корня.

```
(%i22) rho2=subst([exp(-lambda)=1/(1+lambda)], Rho2), ratsimp;
(%o22) rho2= $\frac{\lambda}{\lambda+1}$ 
(%i23) solve(% , lambda);
(%o23) [ lambda =  $-\frac{\rho_2}{\rho_2-1}$  ]
```

Для занижения значения выражения достаточно просто отбросить слагаемое с обратной экспонентой.


```
(%i24) rho2=subst([exp(-lambda)=0], Rho2), ratsimp;
(%o24) rho2= $\frac{\lambda-1}{\lambda}$ 
(%i25) solve(% , lambda);
(%o25) [ lambda =  $-\frac{1}{\rho_2-1}$  ]
```

Построим график — новые кривые должны взять основную в коридор-вилку.



Видим, что правая граница коридора не исходит из нуля координат, что не очень хорошо. Лучше, если бы в тривиальных точках границы сходились, сразу давая корень. Однако главное условие выполняется: при любом значении  $0 \leq \rho^2 < 1$  горизонтальная прямая  $y = \rho^2$  пересечёт как центральную кривую, так и обе границы коридора, так что искомым корень будет взят в «вилку».

Теперь можно вызвать функцию `find_root`. Сначала попробуем провести расчёт при каком-либо одном значении  $\rho^2$ , а если получится — построить график по многим точкам.



В качестве упражнения предлагаем читателю численно решить уравнение Порода в форме

$$\frac{\langle R^2 \rangle}{2l_p^2} = e^{-\lambda} + \lambda - 1$$

при известной левой части ( $0 \div \infty$ ).

### 3.5. Электростатическая персистентная длина

В случае полиэлектролита кулоновское отталкивание между одноимённо заряженными функциональными группами мономерных звеньев приводит к распрямлению цепи, и для её изгиба по персистентному механизму требуется большая энергия по сравнению с незаряженной макромолекулой той же структуры (или со случаем полного экранирования зарядов). Это эквивалентно увеличению эффективной персистентной длины. Величина такого прироста называется электростатической персистентной длиной, к вычислению которой мы и приступаем.

Для незаряженной цепи выполняется равенство

$$\Delta E_0 = \frac{kT}{2} l_{p0} \frac{\theta^2}{s},$$

где  $\Delta E_0$  — прирост потенциальной энергии цепи при изгибе короткого участка длиной  $s$  на угол  $\theta$  (то есть это угол между направлениями концов этого участка),  $l_{p0}$  — персистентная длина незаряженной цепи,  $k$  — постоянная Больцмана,  $T$  — температура. Видим, что  $\Delta E_0 \sim l_{p0}$ . Для такой же деформации макромолекулы полиэлектролита требуется дополнительная энергия  $\Delta E_e$ . Чтобы последнее уравнение оставалось справедливым, необходимо соответственно увеличить  $l_p$ , прибавив к  $l_{p0}$  электростатическую персистентную длину  $l_{pe}$ . Для этого  $\Delta E_e$  должна быть пропорциональна  $kT\theta^2/s$ .

Энергия кулоновского взаимодействия между точечными единичными зарядами  $e$  на расстоянии  $r$  в растворе с учётом ионной атмосферы (дебаевский радиус экранирования  $r_D$ ) равна (СГСЭ):

$$E(r) = \frac{e^2 e^{-r/r_D}}{\epsilon r},$$

$$E(r) := \frac{e^2 \exp(-r/r_D)}{\epsilon r}$$

где  $\epsilon$  — диэлектрическая проницаемость среды. Для распрямлённой цепи  $r$  исчисляется вдоль неё, при дугообразно изогнутой — по хорде, стягивающей эту дугу. Длина хорды окружности, стягивающей дугу длиной  $s$  и видимой из центра под углом  $\theta$  (он же угол между направлениями концов этой дуги — начертите рисунок!), выражается формулой

$$r_{chord} = \frac{2s}{\theta} \sin \frac{\theta}{2}.$$

Поскольку мы рассматриваем малые дуги и малые углы  $\theta$ , при условии чего только и можно аппроксимировать изогнутый участок макромолекулы дугой окружности, произведём разложение в ряд Тейлора по  $\theta$  и ограничимся двумя членами:

```
(%i31) 2*s*sin(theta/2)/theta$
(%i32) taylor(%,theta,0,3);
(%o32)/T/ s -  $\frac{s\theta^2}{24} + \dots$ 
```

Энергию  $\Delta E_e$  найдём суммированием по всем парам точечных зарядов на участке  $s$ :

$$\Delta E_e = \frac{e^2}{\varepsilon} \sum_{i=1}^{M-1} \sum_{j=i+1}^M \left[ \frac{\exp(-r_{ij}/r_D)}{r_{ij}} - \frac{\exp(-(j-i)a/r_D)}{(j-i)a} \right],$$

где  $M$  — полное число звеньев на данном участке цепи,  $r_{ij}$  — расстояние между зарядами звеньев  $i$  и  $j$  в конформации, изогнутой с кривизной  $\theta/s$ . Первое слагаемое соответствует изогнутой конформации, второй — полностью распрямлённой. В данном случае сумма аналитически не вычисляется, поэтому переходим к интегрированию (ср. § 1.3, где применялись оба варианта). При этом для обеспечения точности длина дуги должна быть намного больше дебаевского радиуса:  $s \gg r_D$ .

Для этого введём вместо  $i$  и  $j$  нормированные на единицу переменные  $m$  и  $n$  согласно  $m = ia/s$ ,  $n = ja/s$  ( $0 \leq m < n \leq 1$ ) и заменим двойную сумму двойным интегралом:

$$\Delta E_e = \frac{e^2 s^2}{\varepsilon a^2} \int_0^1 dm \int_m^1 dn \left[ \frac{\exp(-r_{mn}/r_D)}{r_{mn}} - \frac{\exp(-(n-m)s/r_D)}{(n-m)s} \right]$$

Вычисляем хорду  $r_{mn}$ , учитывая, что  $(n-m)$  — это доля длины дуги между соответствующими зарядами (а также доля угла  $\theta$ ):

```
(%i33) subst([s=s*(n-m),theta=theta*(n-m)],%);
(%o33) (n-m) s -  $\frac{(n-m)^3 s \theta^2}{24}$ 
```

Записываем разность энергий — подынтегральное выражение:

```
(%i34) E(%) - E((n-m)*s)$
```

Для аналитического интегрирования такое выражение чересчур сложно. Проверим:

```
(%i35) integrate(%n,m,1);
Is m-1 positive, negative or zero?p;
Is theta positive or negative?p;
Is (m-1)theta-2*sqrt(6) positive, negative or zero?;
```

$$(\%o35) \int_m^1 \frac{e^2 \epsilon e^{\frac{(n-m)^3 s \theta^2}{24} - (n-m)s}}{((n-m)s - \frac{(n-m)^3 s \theta^2}{24})} \frac{e^2 \epsilon e^{-\frac{(n-m)s}{r_D}}}{\epsilon (n-m)s} dn$$

Да, интеграл не взялся. Поэтому воспользуемся малостью  $\theta$  и выделим основной член ряда Тейлора, пропорциональный  $\theta^2$ :

```
(%i36) taylor(%th(2),theta,0,3),factor;
```

$$(\%o36) \frac{e^2 (n-m) \theta^2 (r_D + n s - m s) \epsilon e^{\frac{m s}{r_D} - \frac{n s}{r_D}}}{24 \epsilon s r_D}$$

Берём двойной интеграл:

```
(%i37) integrate(%n,m,1)*s/a$
(%i38) integrate(%m,0,1)*s/a;
```

$$(\%o38) \frac{e^2 \theta^2 \epsilon e^{-\frac{s}{r_D}} ((8 r_D^4 - 3 s r_D^3) \epsilon e^{s/r_D} - 8 r_D^4 - 5 s r_D^3 - s^2 r_D^2)}{24 a^2 \epsilon s^2 r_D}$$

Под экспонентой стоит безразмерное отношение  $s/r_D$  (длина участка цепи, выраженная в дебаевских радиусах). Перейдём от пары параметров  $(s, r_D)$  к паре  $(s, x)$ , где  $x \equiv s/r_D$ :

```
(%i39) x=s/rD$ solve(%rD);
(%o40) [rD=-s/x]
(%i41) subst(%,%th(3)),radcan,expand;
```

$$(\%o41) \frac{e^2 s \theta^2 \epsilon e^{-x}}{24 a^2 \epsilon x} + \frac{5 e^2 s \theta^2 \epsilon e^{-x}}{24 a^2 \epsilon x^2} + \frac{e^2 s \theta^2 \epsilon e^{-x}}{3 a^2 \epsilon x^3} + \frac{e^2 s \theta^2}{8 a^2 \epsilon x^2} - \frac{e^2 s \theta^2}{3 a^2 \epsilon x^3}$$

Примечая общие части слагаемых (что неудивительно, ибо все эти слагаемые должны иметь одну и ту же размерность), изолируем делением безразмерную часть, зависящую только от  $x$ :

```
(%i42) %*24*epsilon*a^2/(s*theta^2*e^2),expand;
```


$$(\%o42) \frac{\epsilon e^{-x}}{x} + \frac{5 \epsilon e^{-x}}{x^2} + \frac{8 \epsilon e^{-x}}{x^3} + \frac{3}{x^2} - \frac{8}{x^3}$$

Сравните с ур. 30.35 из [1]:

$$h(x) = 3x^{-2} - 8x^{-3} + e^{-x}(x^{-1} + 5x^{-2} + 8x^{-3}).$$

Поскольку ранее предположили, что  $s \gg r_D$  ( $x \gg 1$ ), из всех членов полученного выражения следует выбрать наиболее медленно убывающий с ростом  $x$ . Слагаемые с обратной экспонентой не подходят (почему?), из двух

оставшихся выбираем то, у которого меньше степень при  $x$  в знаменателе (он же единственный положительный из двух). Построим график.



Читателю рекомендуется построить графики для всех положительных слагаемых выражения  $h(x)$  и проверить наш выбор.

Раскладываем выражение (до выделения части, зависящей от  $x$ ) на слагаемые и оставляем нужное:

```
(%i44) pickpart(%th(3), 1);
```

```
(%t44) 
$$\frac{e^2 s \theta^2 \varepsilon e^{-x}}{24 a^2 \varepsilon x}$$

```

```
(%t45) 
$$\frac{5 e^2 s \theta^2 \varepsilon e^{-x}}{24 a^2 \varepsilon x^2}$$

```

```
(%t46) 
$$\frac{e^2 s \theta^2 \varepsilon e^{-x}}{3 a^2 \varepsilon x^3}$$

```

```
(%t47) 
$$\frac{e^2 s \theta^2}{8 a^2 \varepsilon x^2}$$

```

```
(%t48) 
$$\frac{e^2 s \theta^2}{3 a^2 \varepsilon x^3}$$

```

```
(%o48) %t48 + %t47 + %t46 + %t45 + %t44
```

Переменная  $x$  своё дело сделала, поэтому возвращаем  $r_D$ :

```
(%i49) subst(%o39, %t47);
```

```
(%o49) 
$$\frac{e^2 \theta^2 r_D^2}{8 a^2 \varepsilon s}$$

```

Напомним, что мы получили, с различными приближениями, изменение электростатической энергии цепи при малом изгибе. Приравниваем его выражению, формально соответствующему механическому изгибу персистентной цепи, и решаем получившееся уравнение:

```
(%i50) %=kT*le*theta^2/(2*s)$
(%i51) solve(% ,le);
(%o51) [ le =  $\frac{e^2 r_D^2}{4 a^2 \epsilon k T}$  ]
```

Сравните итоговое выражение для электростатической персистентной длины с формулой 30.37 из [1]:

$$l_{pe} = \frac{1}{4} \cdot \frac{e^2 r_D^2}{\epsilon a^2 k T}$$

Упражнение. Введите переменную  $x \equiv s/r_D$  до двойного интегрирования и повторите расчёт. Стали ли проще промежуточные выражения, сохранился ли результат?

## 4. Высокоэластичность полимерных материалов

В разделе 2.3 рассчитывалась сила, необходимая для растяжения одиночной свободно-сочленённой цепи, исходя из гауссова распределения. Деформация цепи оказалась прямо пропорциональна растягивающей силе, то есть выполняется закон Гука.

### 4.1. Учёт негауссовости распределения

Однако названное распределение было получено для недеформированной цепи, поэтому такой расчёт верен лишь для весьма малых деформаций. В общем же случае следует учитывать отклонения от гауссовости.

Рассмотрим  $i$ -й сегмент длиной  $a$  свободно-сочленённой цепи, находящийся под углом  $\theta_i$  к произвольно выбранной оси  $OX$  (направлению, вдоль которого будет происходить растяжение). Вне поля действия какой-либо силы все значения этого угла равновероятны и  $\langle \cos \theta_i \rangle = 0$ .

Теперь приложим к концам цепи силу  $\mathbf{f}$  вдоль оси  $OX$ . Эта сила передаётся вдоль по цепи и действует на каждое звено, поэтому всю цепь можно представить как находящуюся в силовом поле, стремящемся ориентировать каждый сегмент вдоль своего направления (аналогия с намагничиванием парамагнетика).

Пусть под воздействием силы  $\mathbf{f}$  радиус-вектор  $\mathbf{R}$  изменится на величину  $d\mathbf{R}$ , при этом внешняя сила совершила работу

$$\delta A = \mathbf{f} d\mathbf{R} = \sum_{i=1}^N \mathbf{f} dr_i.$$

При полевой трактовке внешней силы можно записать

$$\delta A = \sum_i d\varphi_i,$$

где  $d\varphi_i$  – приращение потенциальной энергии  $i$ -го звена. Отсюда

$$\varphi_i = \mathbf{f} \mathbf{r}_i = fl \cos \vartheta_i.$$

Находим статистическую сумму цепи:

$$Z = \int \exp \left[ - \sum_{i=1}^N \frac{fl}{kT} \cos \vartheta_i \right] \prod_{i=1}^N \sin \vartheta_i d\vartheta_i d\varphi_i.$$

В силу одинаковости сегментов

$$Z = \left( \int_0^{\pi} e^{-fa \cos \theta / kT} \sin \theta d\theta \int_0^{2\pi} d\varphi \right)^N.$$

Считаем:

```
(%i1) assume (A>0,beta>0) $
(%i2) exp(-beta*cos(theta))*sin(theta) $
(%i3) integrate(%,theta,0,%pi);
(%o3)  $\frac{e^\beta}{\beta} - \frac{e^{-\beta}}{\beta}$ 
```

Сразу замечаем наличие гиперболического синуса и помогаем Maxima задействовать его:

```
(%i4) %/sinh(beta),exponentialize,factor;
(%o4)  $\frac{2}{\beta}$ 
(%i5) %*sinh(beta);
(%o5)  $\frac{2 \sinh(\beta)}{\beta}$ 
(%i6) (2*%pi*%)^N;
(%o6)  $\frac{\sinh(\beta)^N 4^N \pi^N}{\beta^N}$ 
```

Здесь безразмерная комбинация  $fa/kT$  обозначена за  $\beta$ , по физическому смыслу, это — приведённая сила растяжения. Свободная энергия цепи равна

$$F = -kT \ln Z.$$

```
(%i7) F:-kT*log(%);
(%o7)  $-kT \log\left(\frac{\sinh(\beta)^N 4^N \pi^N}{\beta^N}\right)$ 
```


Средний радиус-вектор  $\langle \mathbf{R} \rangle$ , соответствующий силе  $\mathbf{f}$ , равен

$$\mathbf{R} = -\frac{\partial F}{\partial \mathbf{f}}.$$

Поскольку у нас  $F$  получена как функция от  $\beta$ , дифференцируем её как сложную функцию:

```
(%i8) Rav:-diff(F,beta)*diff(f*a/kT,f),radcan,expand;
(%o8)  $\frac{a \cosh(\beta)^N}{\sinh(\beta)} - \frac{a N}{\beta}$ 
```

Внедрим в выражение гиперболический котангенс и построим график:



Функция  $L(x) = \text{cth } x - 1/x$  называется функцией Ланжевена. Разложим гиперболический котангенс в ряд:

```
(%i11) taylor(coth(beta),beta,0,4);
(%o11)/T/  $\frac{1}{\beta} + \frac{\beta}{3} - \frac{\beta^3}{45} + \dots$ 
```

Видим, что функция Ланжевена игнорирует первый член этого ряда, а при малых  $\beta$  имеем  $L(\beta) \approx \beta/3$ . Подставляя в уравнение для  $\langle \mathbf{R} \rangle$  и разрешая относительно  $f$ , получаем:

```
(%i12) taylor(Rav,beta,0,2);
(%o12)/T/  $\frac{a N \beta}{3} + \dots$ 
(%i13) solve(R=subst([beta=f*a/kT],%),f);
(%o13) [  $f = \frac{3 k T R}{a^2 N}$  ]
```

Сравните с %31 из § 2.3 для гауссовой статистики. Как и следовало ожидать, общая формула переходит в частную при (относительно) малых силах и деформациях.

При больших  $\beta$  можно пренебречь слагаемым  $1/\beta$  и получить аналитическое решение для силы  $f$ :

```

(%i14) pickpart (Rav,1);
(%t14) 
$$\frac{a \cosh(\beta) N}{\sinh(\beta)}$$

(%t15) 
$$\frac{a N}{\beta}$$

(%o15) %t15+%t14
(%i16) solve (R=N*a*coth(beta), beta);
(%o16) [ $\beta = \operatorname{acoth}\left(\frac{R}{aN}\right)$ ]
(%i17) solve (subst ([beta=f*a/kT], first(%)), f);
(%o17) [ $f = \frac{kT \operatorname{acoth}\left(\frac{R}{aN}\right)}{a}$ ]

```

Построение графика для наблюдения выхода среднего радиус-вектора на насыщение оставляем как упражнение читателю.

### 4.2. Деформация полимерной сетки

Рассмотрим образец эластомера, в котором линейные свободно-сочленённые цепи соединены поперечными связями в единую трёхмерную сетку (вспомните вулканизацию каучука). Для простоты примем, что субцепи между точками сшивки имеют одинаковую длину, и в недеформированном состоянии расстояния между узлами сетки (соответствующие расстоянию между концами цепи в несшитом полимере) распределены по закону Гаусса. Ни объём образца, ни его внутренняя энергия при деформировании не меняются. Кроме того, соблюдается принцип аффинности: проекции расстояний между узлами сетки на какую-либо ось пропорциональны изменениям размеров всего образца вдоль этой оси.

Копируем из листинга § 2.2 формулу гауссова распределения с раздельными координатами  $x, y, z$ :

```

(%i18) W: (beta^3*exp(-beta^2*(z^2+y^2+x^2)))/%pi^(3/2);
(%o18) 
$$\frac{\beta^3 e^{-\beta^2(z^2+y^2+x^2)}}{\pi^{3/2}}$$


```

Если обозначить за  $N$  число субцепей в образце, один конец которых находится в начале координат, то число субцепей  $dN$ , у которых второй конец попадёт в объём  $dx dy dz$ , будет равно

$$dN = N \frac{\beta^3}{\pi^{3/2}} e^{-\beta^2(x^2+y^2+z^2)} dx dy dz.$$

Применим формулу Больцмана для энтропии субцепи:

```
(%i19) s0:k*log(W), logexpand=super;
(%o19) k \left( -\beta^2 (z^2 + y^2 + x^2) + 3 \log(\beta) - \frac{3 \log(\pi)}{2} \right)
```

Суммарную энтропию всех субцепей до деформации получим интегрированием

$$S_0 = \int s_0 dN.$$

Хотя интегрировать логично в пределах объёма образца, воспользуемся тем, что длина одиночной субцепи намного меньше любого измерения макроскопического образца, а гауссова экспонента быстро убывает с расстоянием. Поэтому пределы интегрирования можно расширить от минус до плюс бесконечности, что упрощает итоговое выражение:

```
(%i20) integrate(s0*W,x,-inf,inf)$
(%i21) integrate(%,y,-inf,inf)$
(%i22) S0:N*integrate(%,z,-inf,inf),factor;
(%o22) \frac{3(2 \log(\beta) - \log(\pi) - 1) k N}{2}
```

Теперь обратимся к деформации, которую считаем однородной по любому направлению. Если координаты второго конца цепи до деформации были  $\{x, y, z\}$ , то после неё они становятся  $\{\lambda_1 x, \lambda_2 y, \lambda_3 z\}$ , где  $\lambda_1, \lambda_2$  и  $\lambda_3$  задают степени деформации по каждой оси. Очевидно,  $\lambda > 1$  при растяжении и  $< 1$  при сжатии.

Проводим соответствующую замену и получаем энтропию одиночной субцепи после деформации:

```
(%i23) s:subst([x=lambda1*x,y=lambda2*y,z=lambda3*z],s0);
(%o23) k \left( -\beta^2 (\lambda_3^2 z^2 + \lambda_2^2 y^2 + \lambda_1^2 x^2) + 3 \log(\beta) - \frac{3 \log(\pi)}{2} \right)
```

Снова троекратно интегрируем для получения общей энтропии:

```
(%i24) integrate(s*W,x,-inf,inf)$
(%i25) integrate(%,y,-inf,inf)$
(%i26) S:N*integrate(%,z,-inf,inf),factor;
(%o26) \frac{k(\lambda_3^2 + \lambda_2^2 + \lambda_1^2 - 6 \log(\beta) + 3 \log(\pi)) N}{2}
```

Находим изменение энтропии при деформации:

```
(%i27) DS:S-S0,factor;
(%o27) \frac{k(\lambda_3^2 + \lambda_2^2 + \lambda_1^2 - 3) N}{2}
```

Поскольку мы ограничиваемся энтропийной составляющей свободной энергии, то  $\Delta F \approx -T\Delta S$  и

```
(%i28) -T*DS;
```

$$(\%o28) \frac{k(\lambda_3^2 + \lambda_2^2 + \lambda_1^2 - 3)NT}{2}$$

Собираем параметры  $N, k, T$  в единую константу  $G \equiv NkT$ :

```
(%i29) G=N*k*T$ solve(% ,N);
```

$$(\%o30) [N = \frac{G}{kT}]$$

```
(%i31) DF:subst(% ,%th(3));
```

$$(\%o31) \frac{(\lambda_3^2 + \lambda_2^2 + \lambda_1^2 - 3)G}{2}$$

Таким образом, деформационные свойства идеального эластомера определяются числом субцепей (или сшивок, узлов) в сетке и температурой.

### 4.3. Одноосное растяжение

Возьмём частный случай одноосного растяжения. В направлении растяжения (пусть это будет ось  $Ox$ )  $\lambda_1$  обозначим просто за  $\lambda$ , а в двух других направлениях величины  $\lambda_2$  и  $\lambda_3$  будут определяться условием сохранения объёма:  $\lambda_1\lambda_2\lambda_3 = 1$ , откуда

$$\lambda_2 = \lambda_3 = \lambda^{-1/2}.$$

```
(%i32) subst([lambda1=lambda, lambda2=1/sqrt(lambda),
lambda3=1/sqrt(lambda)], DF);
```

$$(\%o32) \frac{G\left(\lambda^2 + \frac{2}{\lambda} - 3\right)}{2}$$

Силу находим, дифференцируя изменение свободной энергии по расстоянию. Здесь производная берётся от сложной функции:

$$f = \frac{\partial \Delta F}{\partial L} = \frac{\partial \Delta F}{\partial \lambda} \cdot \frac{\partial \lambda}{\partial L} = \frac{\partial \Delta F}{\partial \lambda} \cdot \frac{\partial (L/L_0)}{\partial L}$$

```
(%i33) f:diff(% ,lambda)*diff(L/L0,L);
```


$$(\%o33) \frac{G\left(2\lambda - \frac{2}{\lambda^2}\right)}{2L_0}$$

Для расчёта механического напряжения задействуем поперечное сечение в недеформированном состоянии:  $S_0 = V/L_0$  и в деформированном:  $S = V/L = V/L_0 \cdot L_0/L = S_0/\lambda$ .

```
(%i34) sigma:f/(V/(lambda*L0));
```

$$(\%o34) \frac{G\lambda\left(2\lambda - \frac{2}{\lambda^2}\right)}{2V}$$

Строим график относительной силы от относительной деформации ( $\lambda - 1$ ), приняв параметры  $G$  и  $L_0$  за единицу:



Видим, что закон Гука выполняется тем точнее, чем больше  $\lambda$  — теоретическая линия параллельна прямой.

Повторим расчёт с учётом коэффициента Пуассона  $\mu$  — модуля отношения относительных поперечной и продольной деформаций:

$$\mu = -\frac{\lambda_2 - 1}{\lambda_1 - 1}, \quad \lambda_3 = \lambda_2 = 1 - \mu(\lambda_1 - 1).$$

```
(%i36) mu=(1-lambda2)/(lambda-1)$ solve(%, lambda2);
(%o37) [ lambda2=-mu lambda+mu+1 ]
(%i38) subst(append([lambda1=lambda],%, [lambda3=rhs(first(%))]), DF)
(%o38) 
$$\frac{G(2(-\mu \lambda + \mu + 1)^2 + \lambda^2 - 3)}{2}$$

(%i39) f:diff(%, lambda)*diff(L/L0, L), factor;
(%o39) 
$$\frac{G(2\mu^2 \lambda + \lambda - 2\mu^2 - 2\mu)}{L_0}$$

(%i40) sigma:f/(V*(rhs(first(%th(3))))^2/L0), factor;
(%o40) 
$$\frac{G(2\mu^2 \lambda + \lambda - 2\mu^2 - 2\mu)}{V(\mu \lambda - \mu - 1)^2}$$


```

Постройте соответствующие графики, задавшись разумными значениями  $\mu$ . Какое значение этой величины отвечает сохранению объёма при деформации?

#### 4.4. Однородное двумерное растяжение

Пусть образец равномерно растягивается в плоскости XOY, так что  $\lambda_1 = \lambda_2 = \lambda$ , тогда при сохранении объёма  $\lambda_3 = 1/\lambda^2$ . Повторяем расчёт.

```
(%i41) subst ([lambda1=lambda, lambda2=lambda,
              lambda3=1/lambda^2], DF);
              G(2 lambda^2 + 1/lambda^4 - 3)
(%o41) -----
              2
              f:diff(%, lambda)*diff(L/L0, L);
              G(4 lambda - 4/lambda^5)
(%o42) -----
              2 L0
              sigma:f/(V/(lambda*L0));
              G lambda(4 lambda - 4/lambda^5)
(%o43) -----
              2 V
              wxplot2d([f*L0/G, 2*(lambda-1)], [lambda, 1, 10], [nticks, 20])$
```



#### 4.5. Растяжение червеобразной цепи

Без вывода дадим формулу [4]:

$$\tilde{f} \equiv \frac{f l_p}{kT} = \frac{1}{4} \left(1 - \frac{x}{L}\right)^{-2} - \frac{1}{4} + \frac{x}{L} = \frac{1}{4} (1 - \tilde{x})^{-2} - \frac{1}{4} + \tilde{x},$$

где  $\tilde{f}$  — приведённая сила,  $\tilde{x}$  — приведённая деформация ( $0 \leq \tilde{x} < 1$ ).


```
(%i45) FP:1/4*((1-x)^(-2)-1)+x$
```

Для численного решения этого уравнения нужно найти левую и правую границы корня. Поскольку зависимость  $\tilde{f}(\tilde{x})$  монотонно возрастающая, левую границу найдём, поднимая кривую, а правую — опуская её, с сохранением асимптотики и обеспечением аналитического решения получающихся уравнений.

Чтобы поднять кривую, отбросим “+x”, а коэффициент при оставшейся части (вместо 1/4) подберём, исходя из условия равенства производных в нуле (общая касательная):


```
(%i46) C*((1-x)^(-2)-1)$
(%i47) subst([x=0],diff(FP,x))=subst([x=0],diff(% ,x));
(%o47)  $\frac{3}{2}=2C$ 
(%i48) solve(% ,C);
(%o48) [C= $\frac{3}{4}$ ]
(%i49) subst(% ,%th(3));
(%o49)  $\frac{3\left(\frac{1}{(1-x)^2}-1\right)}{4}$ 
```

Чтобы опустить кривую, достаточно просто отбросить “+x”. Строим график:



Как видим, всё в порядке. Теперь осталось решить упрощённые уравнения, выбрать нужный корень (левая и правая границы искомого корня), вызвать функцию `find_root`, проверить её на каком-либо значении и построить график:

```
(%i53) solve(%th(4)=fp,x),radcan;
(%o53) [x=- $\frac{\sqrt{3}\sqrt{4fp+3}-4fp-3}{4fp+3}$ , x= $\frac{\sqrt{3}\sqrt{4fp+3}+4fp+3}{4fp+3}$ ]
(%i54) solve(%th(3)=fp,x);
(%o54) [x=- $\frac{\sqrt{4fp+1}-4fp-1}{4fp+1}$ , x= $\frac{\sqrt{4fp+1}+4fp+1}{4fp+1}$ ]
(%i55) X(fp):=find_root(fp=FP,x,rhs(first('%th(2))),rhs(first('%t53)))$
(%i56) X(20);
(%o56) find_root( $20=x+\frac{1}{(1-x)^2}$ ), x,  $-\frac{1.0(\sqrt{4.0fp+1.0}-4.0fp-1.0)}{4.0fp+1.0}$ , 0.8888888888888888$
(%i57) wxplot2d([X(fp)], [fp,0,20], [nticks,20])$
```



## 5. Уравнение Флори

При взаимодействии с растворителем идеальный гауссов клубок может либо увеличиваться в размерах (набухать), либо уменьшаться (сжиматься). Относительная величина этого эффекта выражается (линейным) коэффициентом набухания  $\alpha$ :

$$\alpha^2 = \frac{s^2}{s_0^2},$$

где  $s^2$  — средний квадрат радиуса инерции клубка,  $s_0^2$  — он же в отсутствие растворителя. Можно также использовать отношение средних квадратов расстояния между концами, которые, как мы видели (§ 1.2), пропорциональны средним квадратам радиуса инерции.

Задача, таким образом, заключается в нахождении величины  $\alpha$  из параметров цепи и свойств её окружения. Для этого цепь рассматривают как термодинамическую систему и выводят её свободную энергию, находя затем минимум последней.

### 5.1. Вывод уравнения Флори

Американский полимерщик П. Флори представил свободную энергию  $F(\alpha)$  полимерного клубка, набухшего в  $\alpha$  раз, в виде суммы двух слагаемых:  $F_{\text{упр}}(\alpha)$ , отвечающего энтропийной упругости цепочки, и  $F_{\text{об}}(\alpha)$ , учитывающего объёмные взаимодействия звеньев.

Свободная энергия идеальной полимерной цепи с заданным значением  $\mathbf{R}$  равна

$$\frac{F(\mathbf{R})}{kT} = \text{const} + \frac{3}{2Na^2} \mathbf{R}^2$$

(прологарифмировано гауссово распределение). При набухании с коэффициентом  $\alpha$  величина  $\mathbf{R}^2$  заменится на  $\alpha^2 \mathbf{R}^2$ . Кроме того,  $\langle \mathbf{R}^2 \rangle = Na^2$ . Вычитание даёт первую составляющую  $F_{\text{упр}}$ :

$$\frac{F_{\text{упр1}}(\alpha)}{kT} = \frac{3}{2} (\alpha^2 - 1).$$

Объём набухшего клубка пропорционален  $\sim \alpha_3 (Na^2)^{3/2}$ , а исходного, ненабухшего  $\sim (Na^2)^{3/2}$ . Энтропия размещения концов цепи в объёме клубка пропорциональна логарифму его объёма. Поэтому изменение этой энтропии при набухании в  $\alpha$  раз будет

$$\ln \alpha^3 N^{3/2} a^3 - \ln N^{3/2} a^3 = 3 \ln \alpha.$$

Таким образом, вторая составляющая  $F_{\text{упр}}$ :

$$\frac{F_{\text{упр2}}(\alpha)}{kT} = 3 \ln \alpha.$$

И в сумме, по Флори,

$$\frac{F_{\text{упр}}(\alpha)}{kT} = \frac{3}{2}(\alpha^2 - 1) - 3 \ln \alpha.$$

Вклад объёмных взаимодействий в свободную энергию цепочки, учитывая малость средней концентрации звеньев в клубке, можно оценить с помощью вириального разложения:

$$\frac{F_{\text{об}}}{kT} = N(nB + n^2C + \dots),$$

где  $n$  — число частиц в единице объёма. Если трактовать клубок как неупорядоченное облако звеньев, распределённых в объёме  $\sim s^3 \sim \alpha^3 N^{3/2} a^3$  с концентрацией  $\sim N/s^3$ , то

$$\frac{F_{\text{об}}}{kT} = s^3 B \left(\frac{N}{s^3}\right)^2 + s^3 C \left(\frac{N}{s^3}\right)^3 + \dots \sim \frac{BN^{1/2}}{a^3} \alpha^{-3} + \frac{C}{a^4} \alpha^{-6} + \dots$$

Таким образом, оставляя только первый член и вводя  $\tilde{B}$ ,

$$\frac{F(\alpha)}{kT} = \frac{3}{2}(\alpha^2 - 1) - 3 \ln \alpha + \frac{\tilde{B}}{\alpha^3}.$$

Минимизируя дифференцированием (проделайте в Maxima), получим:

$$\alpha^5 - \alpha^3 = \tilde{B} = \text{const}.$$

Здесь все условия, определяющие набухание клубка, собраны в константе  $\tilde{B}$ , которая может быть как положительной ( $\alpha > 1$ , набухание), так и отрицательной ( $\alpha < 1$ , сжатие).

При всей простоте уравнения Флори его численное решение (аналитического, в радикалах, не существует — пятый порядок полинома по неизвестному) требует достаточно трудоёмкой подготовки. Изложим её в следующем разделе, чтобы читатель смог в дальнейшем решать и более сложные уравнения.

## 5.2. Численное решение уравнения Флори

Прежде всего, переобозначим:  $\alpha \rightarrow x > 0$ ,  $\tilde{B} \rightarrow A$ . Тогда решению подлежит уравнение (не путать  $F$  со свободной энергией!)

$$F(x, A) = x^5 - x^3 - A = 0.$$

Начнём со знакомства с обрабатываемой функцией «в лицо». Сейчас это несложно — функция  $F(x, A)$  состоит из слагаемого, зависящего только от независимой переменной ( $f(x) = x^5 - x^3$ ), и слагаемого без этой переменной, вобравшего в себя все параметры. Поэтому корень уравнения (нас интересуют только действительные положительные корни) можно представить как абсциссу точки пересечения кривой  $y = f(x)$  с прямой  $y = A$ , параллельной оси абсцисс. Для возврата к прежним обозначениям ( $F(x) = 0$ ) достаточно принять прямую  $y = A$  за ось абсцисс (точнее, сместить график так, чтобы эта прямая совпала с осью абсцисс).

Строим график (рис. 5.1), отмечая на нём пока только экстремумы.




Рис. 5.1: Предварительное знакомство с функцией  $f(x) = x^5 - x^3$

Видим, что при  $A > 0$  уравнение имеет один корень ( $x > 1$ ), а что касается отрицательных значений  $A$ , то, ввиду существования минимума, от нуля до некоего  $A_{\text{мин}}$  имеется два корня (один на интервале  $[0, x_{\text{мин}}]$ , второй на  $[x_{\text{мин}}, 1]$ ), при  $A = A_{\text{мин}}$  корень один ( $x_{\text{мин}}$ ), а при более отрицательных  $A$  корней вообще не существует.

Ищем координаты минимума. Здесь они находятся аналитически:

```
(%i1) x^5-x^3$
(%i2) solve(diff(% , x), x);
(%o2) [x = -sqrt(3)/sqrt(5), x = sqrt(3)/sqrt(5), x = 0]
(%i3) xmin:rhs(second(%));
(%o3) sqrt(3)/sqrt(5)
(%i4) %, float;
(%o4) 0.77459666924148
(%i5) Fmin:subst([x=xmin], %o1), radcan;
(%o5) 2*3^(3/2)/5^(5/2)
(%i6) %, float;
(%o6) -0.18590320061796
```

Таким образом, первое, что должна делать процедура решения уравнения — это отсекал значения  $A$ , меньшие полученного нами  $F_{\text{мин}}$ ; затем проверять, не имеют ли места два тривиальных случая:

$$A = A_{\text{мин}} \rightarrow x = x_{\text{мин}}$$

$$A = 0 \rightarrow x = 1.$$

После этого отсева начинается собственно решение. Отдельно рассмотрим случаи разных знаков  $A$ .

### Случай $A_{\text{мин}} < A < 0$

Уравнение имеет два корня: первый расположен на отрезке  $[0, x_{\text{мин}}]$ , причём

$$F(0) = -A > 0,$$

$$F(x_{\text{мин}}) = A_{\text{мин}} - A < 0;$$

второй же корень находится на отрезке  $[x_{\text{мин}}, 1]$ , при этом

$$F(x_{\text{мин}}) < 0,$$

$$F(1) = -A > 0.$$

В дальнейшем мы предпримем попытку сузить эти границы. Но уже сейчас можно предпринять решение методом половинного деления (дихотомии). Рассмотрим этот метод сначала для второго случая.

### Случай $A > 0$

Действительный положительный корень существует при любом как угодно большом значении  $A$ , поэтому проверки аргумента описанного выше типа не требуется. Выбираем интервал изоляции корня — значения  $x$ , между которыми он заведомо находится, поэтому значения функции в них должны иметь разные знаки. Левая граница этого интервала, очевидно,  $x = 1$ , и  $F(1) = -A < 0$ . Необходимо найти правую границу, на которой  $F$  была бы положительной.

Но сначала поработаем с левой границей — это проще и послужит для тренировки. Нельзя ли как-то сдвинуть её вправо, сузив тем самым интервал изоляции?




Рис. 5.2. Нахождение левой границы (лг) методом сдвига вверх

Посмотрите на график (рис. 5.2). Чтобы кривая пересеклась с прямой  $y = A$  левее корня, эту кривую нужно сдвинуть *вверх*, то есть искусственно увеличить, «завысить» функцию. Перечислим и проиллюстрируем ряд приёмов, применяющихся для этой цели.

1. Простое отбрасывание части выражения (в нашем случае — отрицательного слагаемого, которое, кстати, единственное отрицательное)

$$x^5 - x^3 - A \rightarrow x_{\text{л}}^5 - A = 0.$$

```
(%i7) pickapart(%o1,1);
(%t7) x5
(%t8) -x3
(%o8) %t8+%t7
(%i9) solve(%t7=A,x);
(%o9) [x=%e $\frac{2\pi i}{5}$  A1/5, x=%e $\frac{4\pi i}{5}$  A1/5, x=%e $-\frac{4\pi i}{5}$  A1/5, x=%e $-\frac{2\pi i}{5}$  A1/5, x=A1/5]
```

Нужный нам корень — последний.

2. Замена независимой переменной в минимальной части её вхождений на заведомо меньшее (со знаком минус) или большее (со знаком плюс) значение. Поскольку  $x > 1$ , то единица будет заведомо меньшим корня значением, и мы подставляем её в слагаемое, которое имеет отрицательный знак.

```
(%i10) solve(subst([-%t8=1],%o1)=A,x);
(%o10) [x=%e $\frac{2\pi i}{5}$  (A+1)1/5, x=%e $\frac{4\pi i}{5}$  (A+1)1/5, x=%e $-\frac{4\pi i}{5}$  (A+1)1/5, x=%e $-\frac{2\pi i}{5}$  (A+1)1/5, x=(A+1)1/5]
```

Снова нужный нам корень — последний.

Сравнивая с предыдущим вариантом, видим, что получили «более правую» левую границу, что предпочтительнее. Поэтому отбрасываем вариант (1), оставляя вариант (2).

3. Изменение степени: в нашем случае, поскольку  $x > 1$ , следует увеличивать степень в слагаемом со знаком *плюс* или уменьшать — в слагаемом со знаком *минус*. Уменьшив тройку в степени до нуля, придём к уже рассмотренному второму варианту. Другое упрощение получается при увеличении 5 до 6, возникает квадратное уравнение относительно  $x^3$ .

```
(%i11) subst([%t7=x^6],%o1)=A;
(%o11) x6-x3=A
(%i12) solve(%t7,x);
(%o12) [x= $\frac{(\sqrt{3}i-1)(\sqrt{4A+1}+1)^{1/3}}{2^{4/3}}$ , x= $-\frac{(\sqrt{3}i+1)(\sqrt{4A+1}+1)^{1/3}}{2^{4/3}}$ , x= $\frac{(\sqrt{4A+1}+1)^{1/3}}{2^{1/3}}$ , x= $\frac{(\sqrt{3}i-1)(\sqrt{4A+1}-1)^{1/3}}{2^{4/3}}$ , x= $-\frac{(\sqrt{3}i+1)(\sqrt{4A+1}-1)^{1/3}}{2^{4/3}}$ , x= $-\frac{(\sqrt{4A+1}-1)^{1/3}}{2^{1/3}}$ ]
```

Подходит третий корень.

Можно также уменьшить 3 до 2.5, тогда

$$x = \left( \frac{1 + \sqrt{1 + 4A}}{2} \right)^{2/5}.$$

В обоих случаях в квадратных скобках стоят значения, большие единицы, поэтому предпочтительнее более высокая степень, обеспечивающая «более правую» левую границу. Выбираем поэтому

последний вариант, который, кстати, превосходит и предыдущий (докажите!).

Вернёмся к правой границе. По аналогии делаем вывод, что для её подбора необходимо график сдвинуть вниз. Для этого можно применить уже рассмотренные приёмы, взяв их «с обратным знаком».

1. Простое отбрасывание (единственного положительного слагаемого  $x^5$ ) неприемлемо, так как кривая  $y = -x^3$  и прямая  $y = A$ , проходя, соответственно, ниже и выше оси абсцисс, не пересекаются.

2. Для частичной подстановки заведомо больших/меньших значений немного преобразуем форму записи функции

$$x^5 - x^3 = xx^2(x^2 - 1) = (x^2)^{3/2}(x^2 - 1).$$

Теперь в целях уменьшения с параллельным упрощением можно либо  $x^2$  заменить на  $(x^2 - 1)$ , либо  $x$  принять равным левой границе, то есть числу, заведомо меньшему корня. Переберём эти два варианта.

```
(%i13) assume(A>0)$ solve((x^2-1)^(3/2)*(x^2-1)-A,x);
(%o14) [x=-sqrt(5)*e^(4*i*pi/5)*sqrt(A^(2/5)+1), x=sqrt(5)*e^(4*i*pi/5)*sqrt(A^(2/5)+1), x=-sqrt(5)*e^(2*i*pi/5)*sqrt(A^(2/5)+1), x=sqrt(5)*e^(2*i*pi/5)*sqrt(A^(2/5)+1), x=-sqrt(5)*e^(i*pi/5)*sqrt(A^(2/5)+1), x=sqrt(5)*e^(i*pi/5)*sqrt(A^(2/5)+1), x=-sqrt(5)*e^(-i*pi/5)*sqrt(A^(2/5)+1), x=sqrt(5)*e^(-i*pi/5)*sqrt(A^(2/5)+1), x=-sqrt(5)*e^(-2*i*pi/5)*sqrt(A^(2/5)+1), x=sqrt(5)*e^(-2*i*pi/5)*sqrt(A^(2/5)+1), x=-sqrt(5)*e^(-4*i*pi/5)*sqrt(A^(2/5)+1), x=sqrt(5)*e^(-4*i*pi/5)*sqrt(A^(2/5)+1)]
(%i15) solve(((1+sqrt(1+4*A))/2)^(2/5)*x^2*(x^2-1)-A,x);
(%o15) [x=-sqrt(2)*sqrt((sqrt(4*A+1)+1)^(2/5)+2^(12/5)*A+(sqrt(4*A+1)+1)^(1/5))/sqrt(2)*sqrt(4*A+1)^(1/10), x=sqrt(2)*sqrt((sqrt(4*A+1)+1)^(2/5)+2^(12/5)*A+(sqrt(4*A+1)+1)^(1/5))/sqrt(2)*sqrt(4*A+1)^(1/10), x=-sqrt(2)*sqrt((sqrt(4*A+1)+1)^(2/5)+2^(12/5)*A-(sqrt(4*A+1)+1)^(1/5))/sqrt(2)*sqrt(4*A+1)^(1/10), x=sqrt(2)*sqrt((sqrt(4*A+1)+1)^(2/5)+2^(12/5)*A-(sqrt(4*A+1)+1)^(1/5))/sqrt(2)*sqrt(4*A+1)^(1/10)]
```

3. Уменьшение степени в первом слагаемом (со знаком *плюс*) или увеличение — во втором (со знаком *минус*) с целью получить уравнение степени не выше четвёртой. Технически это можно осуществить так: переписать выражение, разлагая первое или второе слагаемое на множители, а затем заменить на заведомо меньшее корня число  $x$  в положительной степени в первом слагаемом или  $x$  в отрицательной степени — во втором. Всего есть четыре варианта:

$$\boxed{x}x^4 - x^3 - A = 0,$$

$$(x^{5/4})^4 - \boxed{x^{-3/4}}(x^{5/4})^3 - A = 0,$$

$$\boxed{\frac{1}{x^2}} \left(x^{\frac{3}{2}}\right)^3 - \left(x^{\frac{3}{2}}\right)^2 - A = 0,$$

$$(x^{5/3})^3 - \boxed{x^{-1/3}}(x^{5/3})^2 - A = 0.$$

Варианты расположены в порядке убывания абсолютной величины степени  $x$  преобразуемого числа, следовательно, последний предположительно является лучшим. Однако решения уравнений третьей и четвёртой степени громоздки.

**Упражнение.** Сравните все варианты. Какой из них даёт «более левую» правую границу? Предположительно, придётся построить графики зависимости этой границы от  $A$ .

Теперь вернёмся к случаю  $A < 0$  и попытаемся сузить границы локализации корня.




Рис. 5.3. Точки перегиба и вогнутость графика функции  $f(x) = x^5 - x^3$

Из рис. 5.3 видно, что кривую на участке  $[0, 1]$  можно разбить на три монотонных и монотонно вогнутых участка: от нулевой точки до точки перегиба, от точки перегиба до точки минимума, и от точки минимума до точки  $x = 1$ . На каждом из этих участков левая и правая границы корня должны быть разными, поскольку для этих границ берутся более простые функции (полиномы второй и третьей степени), которые неспособны воспроизводить особенности полинома пятой степени. Так, парабола вообще не имеет точек перегиба.

Находим координаты точки перегиба.

```
(%i16) solve(diff(%o1,x,2),x);
(%o16) [x=- $\frac{\sqrt{3}}{\sqrt{10}}$ , x= $\frac{\sqrt{3}}{\sqrt{10}}$ , x=0]
(%i17) xinfl:rhs(second(%));
(%o17)  $\frac{\sqrt{3}}{\sqrt{10}}$ 
(%i18) %,float;
(%o18) 0.54772255750517
(%i19) Finfl:subst([x=xinfl],%o1),radcan;
(%o19)  $\frac{7 \cdot 3^{3/2}}{2^{5/2} 5^{5/2}}$ 
(%i20) %,float;
(%o20) -0.11502173707608
```

Задаём универсальные аппроксимации: параболу и кубическую параболу со свободными коэффициентами.

```
(%i21) QUAD:a*x^2+b*x+c$
(%i22) CUBE:a*x^3+b*x^2+c*x+d$
(%i23) assume(A<0)$
```

Подробно опишем обработку участка кривой от нуля до точки перегиба.

Задаём четыре условия для параболы, два обязательных — значения параболы в крайних точках должны совпадать со значениями «окружаемой» функции, и два необязательных: совпадать должны значения производной.

```
(%i24) subst([x=0],QUAD)=0;
(%o24) c=0
(%i25) subst([x=xinfl],QUAD)=Finfl;
(%o25)  $c + \frac{\sqrt{3} b}{\sqrt{10}} + \frac{3 a}{10} = -\frac{7 \cdot 3^{3/2}}{2^{5/2} 5^{5/2}}$ 
(%i26) subst([x=0],diff(QUAD,x))=0;
(%o26) b=0
(%i27) subst([x=xinfl],QUAD)=subst([x=xinfl],%o1);
(%o27)  $c + \frac{\sqrt{3} b}{\sqrt{10}} + \frac{3 a}{10} = \frac{3^{5/2}}{10^{5/2}} - \frac{3^{3/2}}{10^{3/2}}$ 
```

Поскольку в уравнении параболы три коэффициента, для их расчёта берём два обязательных условия и одно необязательное, получаем уравнение параболы с числовым коэффициентом.

```
(%i28) solve([%th(4),%th(3),%th(2)],[a,b,c]);
(%o28) [[a=- $\frac{7\sqrt{3}}{2^{3/2} 5^{3/2}}$ , b=0, c=0]]
(%i29) subst(first(%),QUAD);
(%o29)  $-\frac{7\sqrt{3} x^2}{2^{3/2} 5^{3/2}}$ 
```

Для кубической параболы обязательными будут все четыре условия.

```
(%i30) subst([x=0], CUBE)=0;
(%o30) d=0

(%i31) subst([x=xinfl], CUBE)=Finfl;
(%o31)  $d + \frac{\sqrt{3}c}{\sqrt{10}} + \frac{3b}{10} + \frac{3^{3/2}a}{10^{3/2}} = -\frac{7 \cdot 3^{3/2}}{2^{5/2} 5^{5/2}}$ 

(%i32) subst([x=0], diff(CUBE, x))=0;
(%o32) c=0


(%i33) subst([x=xinfl], diff(CUBE, x))=subst([x=xinfl], diff(%o1, x))
(%o33)  $c + \frac{2\sqrt{3}b}{\sqrt{10}} + \frac{9a}{10} = -\frac{9}{20}$ 

(%i34) solve([%th(4), %th(3), %th(2), %], [a, b, c, d]);
(%o34) [[a =  $\frac{7\sqrt{2}\sqrt{5}\sqrt{10}-75}{50}$ , b =  $\frac{5 \cdot 3^{3/2}\sqrt{10}-7\sqrt{2} \cdot 3^{3/2}\sqrt{5}}{100}$ , c=0, d=0]]
```

Подставляем полученные значения коэффициентов в уравнение кубической параболы и строим график всех полученных функций.

```
(%i35) subst(first(%), CUBE);
(%o35)  $\frac{(7\sqrt{2}\sqrt{5}\sqrt{10}-75)x^3}{50} + \frac{(5 \cdot 3^{3/2}\sqrt{10}-7\sqrt{2} \cdot 3^{3/2}\sqrt{5})x^2}{100}$ 

(%i36) wxplot2d([%o1, %o29, %, xinfl*x^4-x^3], [x, 0, 0.55], [nticks, 50])
```



За левую границу целесообразно выбрать красную линию: хотя она чуть хуже, чем зелёная, но для неё проще выражение.

```
(%i36) forget(A>0)$ assume(A<0)$ solve(%o28-A,x);
(%o38) [x = - $\frac{2^{3/4} 5^{3/4} \sqrt{-A}}{3^{1/4} \sqrt{7}}$ , x =  $\frac{2^{3/4} 5^{3/4} \sqrt{-A}}{3^{1/4} \sqrt{7}}$ ]
(%i39) rhs(second(%));
(%o39)  $\frac{2^{3/4} 5^{3/4} \sqrt{-A}}{3^{1/4} \sqrt{7}}$ 
```

А сиреневая линия подходит для правой границы.

```
(%i40) solve(xinfl*x^4-x^3-A,x)$
(%i41) limit(% ,A,0),float,ratprint=false;
(%o41) [x = -1.2640284274321854 10-8, x = 1.264028316409883 10-8, x =
1.1102230246251565 10-16, x = 1.825741858350555]
(%i42) rhs(third(%th(2)))$
```

Обратите внимание, что, поскольку это — уравнение четвёртой степени, то выражения для четырёх его корней весьма громоздки, поэтому мы их на экран не выводим, а выбираем нужный корень «вслепую», используя предельное условие: при  $A \rightarrow 0$  граница должна проходить через сам корень, то есть быть нулевой. Этому условию с хорошей точностью удовлетворяет третий корень уравнения, который выделяем, не раскрывая.

Читателю рекомендуется снять значки доллара и посмотреть на результат в полном объёме.

Кратко опишем обработку двух остальных участков.


Второй участок — от точки перегиба до минимума. Условия для параболы.

```
(%i43) subst([x=xinfl],QUAD)=Finfl;
(%o43)  $c + \frac{\sqrt{3} b}{\sqrt{10}} + \frac{3 a}{10} = -\frac{7 3^{3/2}}{2^{5/2} 5^{5/2}}$ 
(%i44) subst([x=xmin],QUAD)=Fmin;
(%o44)  $c + \frac{\sqrt{3} b}{\sqrt{5}} + \frac{3 a}{5} = -\frac{2 3^{3/2}}{5^{5/2}}$ 
(%i45) subst([x=xinfl],diff(QUAD,x))=subst([x=xinfl],diff(%o1,x))
(%o45)  $b + \frac{2\sqrt{3} a}{\sqrt{10}} = -\frac{9}{20}$ 
(%i46) subst([x=xmin],diff(QUAD,x))=subst([x=xmin],diff(%o1,x));
(%o46)  $b + \frac{2\sqrt{3} a}{\sqrt{5}} = 0$ 
```

Здесь задействуем по очереди два необязательных условия.

```
(%i47) solve([%th(4),%th(3),%th(2)], [a,b,c]);
(%o47) [[a =  $\frac{(7 \cdot 2^{3/2} - 17)\sqrt{3}\sqrt{10} + (21\sqrt{2} - 18)\sqrt{3}\sqrt{5}}{100}$ , b =  $\frac{(63\sqrt{2} - 54)\sqrt{5}\sqrt{10} + 105 \cdot 2^{5/2} - 285}{500}$ ,
c =  $\frac{(21\sqrt{2} - 3)\sqrt{3}\sqrt{10} + (21\sqrt{2} - 27)\sqrt{3}\sqrt{5}}{500}$ ]]
(%i48) subst(first(%), QUAD);
(%o48)  $\frac{((7 \cdot 2^{3/2} - 17)\sqrt{3}\sqrt{10} + (21\sqrt{2} - 18)\sqrt{3}\sqrt{5})x^2}{100} - \frac{((63\sqrt{2} - 54)\sqrt{5}\sqrt{10} + 105 \cdot 2^{5/2} - 285)x}{500} + \frac{(21\sqrt{2} - 3)\sqrt{3}\sqrt{10} + (21\sqrt{2} - 27)\sqrt{3}\sqrt{5}}{500}$ 
(%i49) solve([%th(6),%th(5),%th(3)], [a,b,c]);
(%o49) [[a =  $\frac{(7 \cdot 2^{3/2} - 32)\sqrt{3}\sqrt{10} + (21\sqrt{2} - 48)\sqrt{3}\sqrt{5}}{100}$ , b =  $\frac{(21 \cdot 2^{3/2} - 96)\sqrt{5}\sqrt{10} + 315\sqrt{2} - 720}{250}$ ,
c =  $\frac{(21 \cdot 2^{3/2} - 96)\sqrt{3}\sqrt{10} + (63\sqrt{2} - 120)\sqrt{3}\sqrt{5}}{500}$ ]]
(%i50) subst(first(%), QUAD);
(%o50)  $\frac{((7 \cdot 2^{3/2} - 32)\sqrt{3}\sqrt{10} + (21\sqrt{2} - 48)\sqrt{3}\sqrt{5})x^2}{100} + \frac{((21 \cdot 2^{3/2} - 96)\sqrt{5}\sqrt{10} + 315\sqrt{2} - 720)x}{250} + \frac{(21 \cdot 2^{3/2} - 96)\sqrt{3}\sqrt{10} + (63\sqrt{2} - 120)\sqrt{3}\sqrt{5}}{500}$ 
```

Построив график, видим, что нам повезло, и обе параболы охватывают голубую линию решаемого уравнения (кубических парабол не требуется).



Левая граница — зелёная. Обрабатываем её вслепую, выбираем второй корень. 0.5477... — это абсцисса точки перегиба.

```
(%i52) solve(%o50-A, x)$
(%i53) limit(%, A, Finfl), float;
(%o53) [x = 1.001470780977801, x = 0.54772255750517]
(%i54) rhs(second(%th(2)))$
```


Правая граница — красная. Аналогично.

```
(%i55) solve(%o48-A,x)$
(%i56) limit(%,A,Finfl),float;
(%o56) [x=0.54772255750517,x=1.289822461705202]
(%i57) rhs(first(%th(2)))$
```

Третий участок — от точки минимума до единичной точки. Условия для параболы.

```
(%i58) subst([x=xmin],QUAD)=Fmin;
(%o58)  $c + \frac{\sqrt{3}b}{\sqrt{5}} + \frac{3a}{5} = -\frac{2 \cdot 3^{3/2}}{5^{5/2}}$ 
(%i59) subst([x=1],QUAD)=0;
(%o59)  $c + b + a = 0$ 
(%i60) subst([x=xmin],diff(QUAD,x))=0;
(%o60)  $b + \frac{2\sqrt{3}a}{\sqrt{5}} = 0$ 
(%i61) solve([%th(2),%th(3)],[a,b,c]);
(%o61)  $\left[ \left[ a = \frac{4 \cdot 3^{3/2} \sqrt{5} + 45}{25}, b = -\frac{2 \cdot 3^{5/2} \sqrt{5} + 72}{25}, c = \frac{2 \cdot 3^{3/2} \sqrt{5} + 27}{25} \right] \right]$ 
(%i62) subst(first(%),QUAD);
(%o62)  $\frac{(4 \cdot 3^{3/2} \sqrt{5} + 45)x^2}{25} - \frac{(2 \cdot 3^{5/2} \sqrt{5} + 72)x}{25} + \frac{2 \cdot 3^{3/2} \sqrt{5} + 27}{25}$ 
(%i63) subst([x=1],diff(QUAD,x))=subst([x=1],diff(%o1,x));
(%o63)  $b + 2a = 2$ 
(%i64) solve([%th(6),%th(5),%],[a,b,c]);
(%o64)  $\left[ \left[ a = \frac{13\sqrt{3}\sqrt{5} + 80}{25}, b = -\frac{26\sqrt{3}\sqrt{5} + 110}{25}, c = \frac{13\sqrt{3}\sqrt{5} + 30}{25} \right] \right]$ 
(%i65) subst(first(%),QUAD);
(%o65)  $\frac{(13\sqrt{3}\sqrt{5} + 80)x^2}{25} - \frac{(26\sqrt{3}\sqrt{5} + 110)x}{25} + \frac{13\sqrt{3}\sqrt{5} + 30}{25}$ 
```

Построив график, снова видим, что парабол достаточно.



Левая граница — красная.

```
(%i67) solve(%th(5)-A,x)$
(%i68) limit(%A,0),float;
(%o68) [x=0.54919333848297,x=1.0]
(%i69) rhs(second(%th(2)))$
```

Правая граница — зелёная.


```
(%i70) solve(%th(5)-A,x)$
(%i71) limit(%A,0),float;
(%o71) [x=0.61641375809439,x=1.0]
(%i72) rhs(second(%th(2)))$
```

Получив все границы, можно написать подпрограмму-функцию решения уравнения Флори путём вызова стандартной функции find\_root с соответствующими аргументами.

```

(%i73) forget(A<0)$
(%i74) alpha(A):=block([], if A>=0 or A<=Fmin then return ([0,0]),
    if A>Finfl then [find_root(%o1-A,x,'%o39','%o42),find_root(%o1-A,x,'%o69,'%o72)]
    else [find_root(%o1-A,x,'%o54','%o57),find_root(%o1-A,x,'%o69,'%o72)]
) $
(%i75) makelist([A,first(alpha(A))],A,0.999*Fmin,0.001*Fmin,-0.999*Fmin/200)$
(%i76) makelist([A,second(alpha(A))],A,0.999*Fmin,0.001*Fmin,-0.999*Fmin/200)$
(%i77) wxplot2d([discrete,append(%,%t77)]);

```



Сравните с рис. 5.1 или 5.3. Оси координат поменялись местами.

Написание общей подпрограммы-функции, вычисляющей  $\alpha$  и при положительных значениях  $A$ , оставляем заключительным упражнением читателю.

## Литература

1. Гросберг А.Ю., Хохлов А.Р. Статистическая физика макромолекул: учеб. руководство. — М.: Наука, 1989. — 344 с.
2. Цветков В.Н. Жесткоцепные макромолекулы. — Л.: Наука, 1986. — 380 с.
3. Кленин В.И., Федусенко И.В. Высокомолекулярные соединения: учебник для студентов хим. фак. — Саратов: Изд-во Сарат. ун-та, 2008. — 440 с.
4. Marko J.F., Siggia E.D. Stretching DNA // *Macromolecules*. 1995. Vol. 28, No. 26. P. 8759–8770.
5. Petrosyan R. Improved approximations for some polymer extension models // *Rheol. Acta*. 2017. Vol. 56, No 21.

## Оглавление

Предисловие .....	3
1. Средние величины в моделях идеальной полимерной цепи.....	5
1.1. Свободно-сочленённая цепь.....	5
1.2. Средний радиус инерции .....	6
1.3. Цепь с фиксированным валентным углом .....	7
1.4. Цепь с заторможенным внутренним вращением .....	8
1.5. Сегмент Куна.....	8
1.6. Предельный переход.....	9
2. Функции распределения в моделях идеальной полимерной цепи .....	14
2.1. Задача складного метра.....	14
2.2. Функция распределения вектора между концами цепи .....	16
2.3. Энтропийная эластичность одиночной цепи .....	18
2.4. Функция распределения расстояния между концами цепи .....	19
2.5. Свойства распределения расстояния между концами цепи.....	20
2.5.1. Положение максимума (наиболее вероятное расстояние между концами цепи).....	20
2.5.2. Среднее расстояние между концами цепи.....	20
2.5.3. Средний квадрат расстояния между концами цепи .....	20
2.5.4. Средняя флуктуация квадрата расстояния между концами цепи .....	21
2.6. Гидродинамический радиус клубка .....	22
2.7. Случай кольцевой макромолекулы .....	23
3. Модель червеобразной цепи. Персистентная длина .....	26
3.1. Корреляция между положениями сегментов.....	26
3.2. Проекция цепи на направление первого сегмента .....	27
3.3. Средний квадрат длины .....	28
3.4. Альтернативная форма уравнения Порода. Его численное решение .....	29
3.5. Электростатическая персистентная длина.....	32
4. Высокоэластичность полимерных материалов .....	37
4.1. Учёт негауссовости распределения .....	37
4.2. Деформация полимерной сетки .....	40
4.3. Одноосное растяжение.....	42
4.4. Однородное двумерное растяжение .....	44
4.5. Растяжение червеобразной цепи .....	44
5. Уравнение Флори .....	47
5.1. Вывод уравнения Флори.....	47

5.2. Численное решение уравнения Флори .....	48
Литература .....	61

САРАТОВСКИЙ ГОСУДАРСТВЕННЫЙ УНИВЕРСИТЕТ ИМЕНИ Н. Г. ЧЕРНЫШЕВСКОГО